
1. はじめに 

 

WPSは DB2, MySQL, Netezza, Oracle, PostgreSQL, SQL Server, Teradata などの商用、ま

たはオープンソースのデータベース（以下 DBとも表記）に対するアクセスインターフェー

ス機能をサポートしています。インターフェース機能に含まれるのは、データの読み書き、

テーブルの作成・編集・削除を含みます。サポートするインタフェース方式は、DBごとに

固有のコマンドライン接続方式（CLIドライバー、ネイティブドライバー）に加えて、多く

の DB に対してそれぞれの DB ベンダーから提供されているオープンデータベース接続方

式（ODBCドライバー）の両方です。 

 

本資料は 64ビット版Windows8.1上で動作する以下のデータベースプロダクトをローカル

環境にインストールし、各プロダクトに付随しているサンプルデータベースを例として CLI

ドライバー経由および ODBC ドライバ経由で WPS3.1.1 からそれぞれの DB テーブルの入

出力を行う場合を例示し、WPSの DBインタフェース機能の使い方を説明したものです。 

 

DB2 Express-C 10.5（SAMPLEデータベース） 

MySQL 5.6（sakilaデータベース（スキーマ）） 

ORACLE Express Edition 11.2（scottスキーマ） 

PostgreSQL 9.3.5（dvdrentalデータベース） 

SQL Server 2012 Express Edition（AdventureWorks2012データベース） 

 

なお、本資料では、主に SHIFT-JISエンコード（MS932エンコード）の日本語データを取

扱っています。そのため、WPSワークベンチの設定が、以下のとおり設定されていること

を前提としています。 

 

・起動オプションに encoding="SHIFT-JIS" 1 

・ワークスペースのテキストファイルエンコードに MS9322 

                                                   
1 「WPSサーバーエクスプローラー」ビューの「ローカルサーバー」を右クリック→「プロパティ(R)」をクリック 

→ 「プロパティ：ローカルサーバー」画面の左領域の「起動オプション」をクリック → 右領域の「追加」をクリッ

ク → 「起動オプション」画面出現 →「選択」をクリック → 出現した「起動オプションを選択」画面のオプション

リストの中から「ENCODING」を探してクリック → 「OK」をクリック → 「起動オプション」画面の「名前：」欄

に ENCODING が入っていることを確認 → 「値：」欄に SHIFT-JISを入力 → 「OK」。なお、バージョン 3.1.1か

ら LOCALE="JAPANESE_JAPAN"を指定できるようになり、この指定を行うと、ENCODING="SHIFT-JIS" に自動設定

されるようになりました。 

 

2 「メニューバー」の「ウィンドウ(W)」→ 「設定(P)」→ 左領域の「一般」→ 「ワークスペース」をクリック → 

右領域の左下部にある「テキスト・ファイル・エンコード(T)」の「その他(O)」ラジオボタンをクリック → 入力欄に 

MS932 とキー入力 → 「OK」。なお、バージョン 3.1.1では既定の設定になりました。 

プログラムをバッチ実行する場合は、-encoding SHIFT-JIS （起動オプション）を指定して

実行することが必要です。 

 

［WPS がサポートしているデータベース］ 

 

WPS3.1.1でアクセスインタフェースがサポートされているDBとそのインターフェース機

能の種類とアクセスの種類は以下のとおりです。 

 

 

 

OS別サポート状況は以下のとおりです。 

 

データベース
読み取

り
書き込

み
更新

新規
テーブ
ルの作

成

パスス
ルー

（暗黙）

パスス
ルー

（明示）

バルクアッ
プロード

Action Matrix
DB2  (z/OS以外)

Greenplum
Informix
Kognitio
MySQL
Netezza
ODBC  (SQL Serverのみ)

OLEDB
Oracle
PostgreSQL
SAND
SQL Server
Sybase
Teradata
Vertica

(インタフェース機能)

バルクロード

データベース
AIX

pSeries(P
ower)

Linux
x86

Mac OS
X

Solaris
SPARC

Solaris
x86

Windows
x86

Linux
System z

z/OS
System z

Action Matrix
DB2

Greenplum
Informix
Kognitio
MySQL
Netezza
ODBC
OLEDB
Oracle

PostgreSQL
SAND

SQL Server
Sybase

Teradata
Vertica

(OS環境)

※ ODBC接続についてはOSや対象DBによってサードパーティ製品の導入が必要になる

場合もあります。 

 

なお、ここに示した例は、ローカル環境にインストールした DBに管理者ユーザーでログイ

ンできる環境で実行しています。ネットワーク環境や、セキュリティ環境の違いにより、こ

こで示したプログラムや設定では DB アクセスが実行できない場合もあります。その場合

は、DBアクセス環境等を確認の上、必要な修正を加えて実行してください。 

 

2. DB2 とのインタフェース 

 

［SAMPLE サンプル DB］ 

 

DB2 Express-C 10.5 に付随する sampleデータベースを例として用います。 

 

 

 

2.1. ネイティブドライバー経由の DB テーブルの入出力 

 

まず、コマンドラインインタフェース（CLIドライバー、ネイティブドライバー）を用いて

WPSから DBテーブルの入出力を行ってみましょう。 

ネイティブドライバーは DB と一緒にインストールされており、一般に ODBC ドライバー

より高速に DBアクセスができる点でメリットがあると言われています。 

WPSの DB2エンジンはネイティブドライバーを経由して DB2データベースとアクセスを

行う仕組みを提供します。 

［SQL 言語を使用した DB 読み取り］ 

 

DB へのアクセスは SQL 言語を用いて行うことが多く、WPS でも SQL プロシジャを用い

て SQL言語による DBアクセスが可能です。 

 

SQL言語を用いてWPSからネイティブドライバー経由でDB2 の DBアクセスを行う場合

は、以下のように、エンジン名として DB2 を指定し、CONNECT 文のカッコ内に 

datasrc=DB名 などの接続オプションや schema=スキーマ名、dbmax_text=n（文字テキス

トの読み取り最大長さの設定）などの libnameオプションを必要に応じて指定します。 

 

以下は sampleデータベースの staffテーブルから、DEPT=10または 20に該当するレコー

ドのみ抽出した全カラムを含むWPSデータセットWORK.staffを作成しています。 

 

proc sql; 
  connect to DB2 (datasrc=sample); 
  create table staff as 
  select * 
  from connection to DB2 
  ( 
    select * 
    from staff 
    where DEPT in (10, 15); 
  ); 
  disconnect from DB2; 
quit; 

※ 3行目の create table 文はWPSデータセット WORK.staffを作成する指定です。 

8 行目の from 節で指定した staff は sample データベースの staff テーブルを参照してい

ます。（データベースのテーブル名の指定は、本来、「スキーマ名.テーブル名」の形式で指

定します。しかし、テーブルの所属するスキーマ名が接続する DB2 ユーザー名と同じ場合

は「スキーマ名.」の部分は省略できるため、ここでは省略しています。） 

 

(WPSワークベンチの実行ログ) 

8         proc sql; 
9           connect to DB2 (datasrc=sample); 
NOTE: Successfully connected to database DB2 as alias DB2. 
10          create table staff as 
11          select * 
12          from connection to DB2 
13          ( 
14            select * 
15            from staff 
16            where DEPT in (10, 15); 
17          ); 
NOTE: Data set "WORK.staff" has 8 observation(s) and 7 variable(s) 
18          disconnect from DB2; 
NOTE: Successfully disconnected from database DB2. 
19        quit; 

NOTE: Procedure sql step took : 
      real time : 1.594 
      cpu time  : 0.031 

 

作成されたWPSデータセットWORK.staffデータセットの内容を確認します。 

 

 

※ 注意：DB2ドライバーによる DBへのアクセス制限等の設定によっては、SQLプロシジ

ャの CONNECT 文の（）内のオプションに、その他の指定が必要となる場合があります。

指定可能なオプションは、以下のように、ヘルプビューで確認できます。 

 

まず、ヘルプビュー3の左上にある「内容」をクリックします。ヘルプ全体の目次が表示さ

れますので、WPS Reference for Language Elements をクリックします。 

 

WPSレファレンスを展開し、WPS Engine for DB2 を選択します。 

                                                   
3 ヘルプビューが存在しない場合は、「ウィンドウ(W)」→「ビューの表示(V)」→ 「その他(O)」→「ヘルプ」フォル

ダー → 「ヘルプ」からビューを出現させます。また、「ヘルプ(H)」→「ヘルプ目次(H)」から別ウィンドウで表示さ

れるメニューでも同じです。 

Library Engines → DB2を選択します。（z/OS版4の場合は DB2_zOSを選択） 

 

 

データベース接続オプション（connection option）が表示されます。 

                                                   
4 z/OS版では procedursにあるデータ抽出用の DB2EXTプロシジャも利用可能です。 

これらの接続オプション（user= password= datasrc= noprompt= prompt= required=）は DB2

ドライバー経由でのデータベースアクセスの際の CONNECT 文の DBMS オプション指定

としても、また、次に節で説明する libnameステートメントによる DB接続オプションとし

ても用いることができ、必要に応じて指定します。 

 

［libname ステートメントによる DB 入力］ 

 

WPSのエディタから以下のプログラムを入力し、実行します。 

 

libname in DB2 datasrc=sample; 

 

この指定は、WPS がデータベース sample に DB2 ネイティブドライバー経由で接続し、

sample データベースのテーブルとビューをライブラリ名 in で参照できるようにする指定

です。（入出力両方可能） ただし、DBへの接続やデータテーブルの検索に必要な権限の設

定や、他の接続オプションが必要となる場合があります。 

 

さらに、libname ステートメントには schema=スキーマ名、dbmax_text=n（文字テキスト

の読み取り最大長さの設定）のような libnameオプションを追加指定できます。以下のよう

な libname オプションが指定可能です。（前出の DB2 エンジンのヘルプ画面から確認でき

ます。）なお、schema=スキーマ名を指定しない場合の既定のスキーマ名は DB 接続ユーザ

ー名です。 

 

 

 

さて、実行すると、ログに 「ライブラリ名 in が割り当てられた」というメッセージが表

示されます。このメッセージは DB2エンジン経由で DBと接続できたことを表します。 

29        libname in DB2 datasrc=sample; 
NOTE: Library in assigned as follows: 
      Engine:        DB2 
      Physical Name: sample 

 

実行後、WPS サーバーエクスプローラービュー内の ローカルサーバー の下の Libraries 

を展開し、inライブラリがあることを確認してください。この中に sampleデータベースの

テーブル名およびビュー名が含まれていることがわかります。 

 

※ 右矢印がついたアイテムはテーブル名（展開すると、テーブルに含まれる項目名と項目

タイプを表すアイコンが表示されます。）、付いていないのはビュー名です。 

 

これらのテーブルやビューは、WPSデータセットに対する操作と同じファイル操作が可能

です。例えば、以下のように、テーブル名を右クリックして出現するメニューの「プロパテ

ィ」を選択すると、カラム名などを確認することができます。 

※ 【重要な注意】：テーブル名をダブルクリックすると、データセットビューアが起動し、

テーブルの中身を表示しようとします。この操作はテーブルのサイズが大きいときはWPS

セッションが動かなくなったり不意に終了してしまうなどの危険を伴いますので、できる

だけ避けてください。 

 

さて、DB2 エンジンを libname ステートメントで指定した後は、以下のプログラムのよう

に、DB テーブルやビューをWPS データセットに変換することなく、直接プロシジャ入力

に用いることもできます。この例では、staffテーブルを入力として職種（JOB）別の平均給

料（SALARY）の棒グラフを描いています。 

 

goptions cback=white; 
title "Salary mean by Job";  
proc gchart data=in.staff; 
  vbar job/sumvar=salary type=mean patternid=midpoint width=10 space=5 mean; 
run; 

※ 注意：グラフィック表示はWPS3.1.1では日本語の表示はサポートされていません。 

 

また、以下のように、SAS 言語の DATA ステップでも DB テーブルを読むことができるよ

うになります。また、SQLプロシジャを使用して処理することも可能ですので、libnameス

テートメントを使って DBアクセスを行うときは、SAS言語と SQL言語いずれも使用可能

になるというメリットがあります。 

 

下の例は、staffテーブルから職種（JOB）が"Mgr"か"Sales"のいずれかに該当するレコード

を選択した変数Name, Job, Salaryを含むWPSデータセットWORK.Mgr_or_SalesをDATA

ステップで作成しています。 

 

data Mgr_or_Sales; 
  set in.staff(keep=Name Job Salary); 
  where Job in ('Mgr', 'Sales'); 
run; 

 

［WPS データセットの DB への出力方法］ 

 

以下のいずれかの方法でWPSデータセットを DBのテーブルへ出力できます。 

 

① DBLOADプロシジャ 

② SQLプロシジャ 

③ DATAステップ 

 

①の方法では DB への接続方法を DBLOAD プロシジャで指定しますので、libname ステー

トメントで DBエンジンを指定する必要はありません。しかし、②と③の方法では、事前に

DB への接続に用いるエンジン名と接続先（DB 名やスキーマ名）を libname ステートメン

トで指定しておく必要があります。 

 

① DBLOADプロシジャ 

 

前節の最後のプログラムで sample テーブルから職種が "Mgr" か "Sales" に該当するレコ

ードを選択したWPS データセットWORK.Mgr_or_Sales を DB2 の Mgr_or_Sales テーブ

ルへ出力してみましょう。以下のようにプログラムを書いて実行します。 

 

proc dbload data=Mgr_or_Sales dbms=DB2; 
  datasrc=sample; 
  table=Mgr_or_Sales; 
  load; 
run;quit; 

※ 注意： DB にアクセスするため DB2 エンジンの接続オプションや libname オプション

がさらに必要な場合は、loadステートメントの前に、オプション名=値; の形式で指定を追

加してください。 

 

② SQLプロシジャによるテーブルへの読み書き 

 

Libnameステートメントでエンジン名を指定すると、DBテーブルをWPSデータセットと

同じように扱えるようになります。 

以下の例は sample データベースの sales テーブルを読んで、SALES_PERSON="LEE"に

該当するレコードのみ抽出した全カラムを含むテーブル LEEを作成しています。 

 

libname sales DB2 datasrc=sample; 
proc sql; 
  create table sales.LEE as 
  select * 
  from sales.sales 
  where sales_person = "LEE"; 
quit; 

※ この例では、Sales ライブラリは DB2 の sample データベースを参照するよう最初の

libname ステートメントで定義しています。したがって、SQL プロシジャの中の Sales ラ

イブラリは WPS データライブラリではなく、sample DB を表すものと解釈され、読み取

りも書き出しも DBテーブルが対象になります。 

 

③ DATAステップ 

以下の例は②の SQL プロシジャを使った処理と同じことを SAS 言語の DATA ステッププ

ログラムで行っています。 

 

libname sales DB2 datasrc=sample; 

/* 同じ名前のテーブルを一旦削除 */ 
proc datasets lib=sales; 
  delete LEE; 
run;quit;   

/* DATAステップでDBテーブルを読み書き */ 
data sales.LEE; 
  set sales.sales; 
  where sales_person = "LEE"; 
run;   

 

2.2. ODBC ドライバー経由の DB テーブルの入出力 

 

ODBC ドライバーは、Windows 環境や Linux 環境から各種 DB とのインタフェースを共通

方式で行うためのソフトウェアです。ほとんどの DB の ODBC ドライバーが DB ベンダや

サードパーティから提供されており、ODBCドライバーの設定を行えば、DBごとの細かい

アクセス条件等の設定を気にせずに、共通の構文で DBアクセスが実現できるという利点が

あります。 

 

［ODBC ドライバーのインストール］ 

 

DB2のODBCドライバーはDB2インストール時に自動でインストールされます。 もしも、

次の［PC側の ODBC ドライバーの設定］で DB2の ODBCドライバーが見当たらない場合

は、インストール環境を確認してください。 

 

［PC 側の ODBC ドライバーの設定］ 

 

WPSが導入されている PC側では、以下のような手順で、DB2 の sampleデータベースを

ODBCデータソースとして登録しておきます。 

 

コントロールパネル → システムとセキュリティ → 管理ツールで ODBC データソースを

選択します。 

（左図）「追加」をクリックします。 

（右図）リストにある DB2用 ODBCドライバーを選択し、「完了」を押します。 

 

 

（左図）新規データソースの名前と説明にはユーザーが自由に入力できます。ここでは、名

前に「DB2_SAMPLE」、説明には「DB2 Express-C sample database」と入力し「OK」を押

します。 

（右図）ODBCデータソースリストに DB2用 ODBCドライバー「DB2_SAMPLE」が追加

されたことを確認し、これを選択してから「構成」を押します。 

 

 

「接続」をクリックし接続テストを行います。 

正常に接続できたなら「OK」を押し ODBCドライバーの設定を終了します。 

 

［SQL 言語を使用した DB 読み取り］ 

 

SQL言語を用いてWPSからODBCドライバー経由でDB2 のDBアクセスを行う場合は、

以下のように、エンジン名として ODBC を指定し、CONNECT 文のカッコ内に 

datasrc=ODBC データソース名 などの接続オプションや dbmax_text=n（文字テキストの

読み取り最大長さの設定）などの libnameオプションを必要に応じて指定します。 

 

下記は、ODBCドライバーを用いて、sampleデータベースの staffテーブルから、DEPT=10

または 20に該当するレコードのみ抽出した全カラムを含むWPSデータセットWORK.staff

を作成する例となっています。 

 

proc sql; 
  connect to ODBC (datasrc=DB2_SAMPLE); 
  create table staff as 
  select * 
  from connection to ODBC 
  ( 
    select * 
    from staff 
    where DEPT in (10, 15); 
  ); 
  disconnect from ODBC; 
quit; 

 

(WPSワークベンチの実行ログ) 

314       proc sql; 
315         connect to ODBC (datasrc=DB2_SAMPLE); 
NOTE: Successfully connected to database ODBC as alias ODBC. 
316         create table staff as 
317         select * 
318         from connection to ODBC 
319         ( 
320           select * 
321           from staff 
322           where DEPT in (10, 15); 

323         ); 
NOTE: Data set "WORK.staff" has 8 observation(s) and 7 variable(s) 
324         disconnect from ODBC; 
NOTE: Successfully disconnected from database ODBC. 
325       quit; 
NOTE: Procedure sql step took : 
      real time : 0.359 
      cpu time  : 0.015 

※ ODBCドライバーによる DBへのアクセス制限等の設定によっては、SQLプロシジャの

CONNECT文の（）内のオプションに、その他の指定が必要となる場合があります。以下の

ようにヘルプビューで指定可能なオプションが確認できます。 

 

ヘルプビュー5の左上にある「内容」をクリックします。 

ヘルプ全体の目次が表示されます。WPS Reference for Language Elements をクリックし

ます。 

 

 

WPSレファレンスを展開し、WPS Engine for ODBC を選択します。 

 

 

                                                   
5 ヘルプビューが存在しない場合は、「ウィンドウ(W)」→「ビューの表示(V)」→ 「その他(O)」→「ヘルプ」フォル

ダー → 「ヘルプ」からビューを出現させます。また、「ヘルプ(H)」→「ヘルプ目次(H)」から別ウィンドウで表示さ

れるメニューでも同じです。 

その後、展開を何度か行うと接続オプション（connection option） が表示されます。 

 

 

これらのオプション（user= password= datasrc= noprompt= prompt= readbuff= required=）

は ODBC ドライバー経由でのデータベースアクセスの際の CONNECT 文の DBMS オプシ

ョン指定としても、また、次に説明する libnameステートメントによるデータベースアクセ

スの接続オプションとしても用いることができ、必要に応じて指定します。 

 

［libname ステートメントによる DB 入力］ 

 

WPSのエディタから以下のプログラムを実行します。 

 

libname in ODBC datasrc=DB2_SAMPLE; 

 

この指定は、WPSがデータソース DB2_SAMPLEに ODBCドライバー経由で接続し、ライ

ブラリ名 inで参照できるようにする指定です。（入出力両方可能） 

ただし、DBへの接続やデータテーブルの検索に必要な権限の設定や、他の接続オプション

が必要となる場合があります。 

 

さらに、libname ステートメントには dbmax_text=n（文字テキストの読み取り最大長さの

設定）などの libname オプションを追加指定できます。指定可能な libname オプションは

DB2エンジンの libnameオプションとほぼ同じですが、詳細は前出の ODBC エンジンのヘ

ルプ画面から確認してください。（Connection-optionの下に libname-optionが記載されてい

ます。） 

 

実行すると、ログに「ライブラリ名 in が割り当てられた」というメッセージが表示されま

す。このメッセージは ODBC経由で DBと接続できたことを表します。 

 

345       libname in ODBC datasrc=DB2_sample; 
NOTE: Library in assigned as follows: 
      Engine:        ODBC 
      Physical Name: DB2_sample 

 

実行後、WPS サーバーエクスプローラービュー内の ローカルサーバー の下の Libraries 

を展開し、inライブラリがあることを確認してください。この中に sampleデータベースの

テーブル名およびビュー名が含まれていることがわかります。 

 

※ 右矢印がついたアイテムはテーブル名（展開すると、テーブルに含まれる項目名と項目

タイプを表すアイコンが表示されます。）、付いていないのはビュー名です。 

 

これらのテーブルやビューは、WPSデータセットに対する操作と同じファイル操作が可能

です。例えば、以下のように、テーブル名を右クリックして出現するメニューの「プロパテ

ィ」を選択すると、カラム名などを確認することができます。 

※ 【重要な注意】：テーブル名をダブルクリックすると、データセットビューアが起動し、

テーブルの中身を表示しようとします。この操作はテーブルのサイズが大きいときはWPS

セッションが動かなくなったり不意に終了してしまうなどの危険を伴いますので、できる

だけ避けてください。 

 

さて、ODBC エンジンを libname ステートメントで指定した後は、以下のプログラムのよ

うに、DB テーブルやビューをWPS データセットに変換することなく、直接プロシジャ入

力に用いることもできます。この例では、staffテーブルを入力として職種（JOB）別の平均

給料（SALARY）の棒グラフを描いています。 

 

goptions cback=white; 
title "Salary mean by Job";  
proc gchart data=in.staff; 
  vbar job/sumvar=salary type=mean patternid=midpoint width=10 space=5 mean; 
run; 

※ 注意：グラフィック表示はWPS3.1.1では日本語の表示はサポートされていません。 

 

また、以下のように、SAS 言語の DATA ステップでも DB テーブルを読むことができるよ

うになります。また、SQLプロシジャを使用して処理することも可能ですので、libnameス

テートメントを使って DBアクセスを行うときは、SAS言語と SQL言語いずれも使用可能

になるというメリットがあります。 

 

下の例は、staffテーブルから職種（JOB）が"Mgr"か"Sales"のいずれかに該当するレコード

を選択した変数Name, Job, Salaryを含むWPSデータセットWORK.Mgr_or_SalesをDATA

ステップで作成しています。 

 

data Mgr_or_Sales; 
  set in.staff(keep=Name Job Salary); 
  where Job in ('Mgr', 'Sales'); 
run; 

 

［WPS データセットの DB への出力方法］ 

 

以下のいずれかの方法でWPSデータセットを DBのテーブルへ出力できます。 

 

① DBLOADプロシジャ 

② SQLプロシジャ 

③ DATAステップ 

 

①の方法では DB への接続方法を DBLOAD プロシジャで指定しますので、libname ステー

トメントで DBエンジンを指定する必要はありません。しかし、②と③の方法では、事前に

DB への接続に用いるエンジン名と接続先（DB 名やスキーマ名）を libname ステートメン

トで指定しておく必要があります。 

 

① DBLOADプロシジャ 

 

前節の最後のプログラムで sample テーブルから職種が "Mgr" か "Sales" に該当するレコ

ードを選択したWPS データセットWORK.Mgr_or_Sales を DB2 の Mgr_or_Sales テーブ

ルへ出力してみましょう。以下のようにプログラムを書いて実行します。 

 

proc dbload data=Mgr_or_Sales dbms=ODBC; 
  datasrc=DB2_SAMPLE; 
  table=Mgr_or_Sales; 
  load; 
run;quit; 

※ 注意： DBにアクセスするための権限や ODBCエンジンの接続オプションや libnameオ

プションがさらに必要な場合は、loadステートメントの前に、オプション名=値; の形式で

指定を追加してください。 

 

② SQLプロシジャによるテーブルへの読み書き 

 

Libnameステートメントでエンジン名を指定すると、DBテーブルをWPSデータセットと

同じように扱えるようになります。 

以下の例は sample データベースの sales テーブルを読んで、SALES_PERSON="LEE"に

該当するレコードのみ抽出した全カラムを含むテーブル LEEを作成しています。 

 

libname sales ODBC datasrc=DB2_SAMPLE; 
proc sql; 
  create table sales.LEE as 
  select * 
  from sales.sales 
  where sales_person = "LEE"; 
quit; 

※ この例では、Salesライブラリは ODBCの DB2_SAMPLEデータソースを参照するよう

最初の libname ステートメントで定義しています。したがって、SQL プロシジャの中の

Salesライブラリは WPSデータライブラリではなく、sample DBを表すものと解釈され、

読み取りも書き出しも DBテーブルが対象になります。 

 

③ DATAステップ 

以下の例は②の SQL プロシジャを使った処理と同じことを SAS 言語の DATA ステッププ

ログラムで行っています。 

 

libname sales ODBC datasrc=DB2_SAMPLE; 

/* 同じ名前のテーブルを一旦削除 */ 
proc sql; 
  drop table sales.LEE; 
quit;   

/* DATAステップでDBテーブルを読み書き */ 
data sales.LEE; 
  set sales.sales; 
  where sales_person = "LEE"; 
run;   

 

3. MySQL とのインタフェース 

 

［sakila サンプル DB］ 

 

MySQL Community Server 5.6 に付随する sakilaデータベースを例として用います。 

 

 

 

3.1. ネイティブドライバー経由の DB テーブルの入出力 

 

まず、コマンドラインインタフェース（CLIドライバー、ネイティブドライバー）を用いて



25 
 

WPSから DBテーブルの入出力を行ってみましょう。 

 

［ネイティブドライバのインストールと Path の定義］ 

 

WPSは MySQLとの接続を行うネイティブドライバーとして、Connector/C(libmysqlclient) 

ライブラリを用います。このファイル（libmysqlclient.dll）はインストールすると、MySQL

インストールディレクトリの下の libディレクトリ内に保管されます。これをアクセスでき

るように、以下のように、環境変数 Path に定義しておく必要があります。 

 

（左図）コントロールパネル → システム → システムの詳細設定を開きます。 

「環境変数」をクリック → 「Path」をクリックします。 

（右図）ユーザー環境変数 Pathを選択し、「編集」をクリックします。なお、他のユーザー

でも有効な Pathを設定したい場合はシステム環境変数の Pathを選択し編集してください。 

 

 

変数値の最後に、以下のように、MySQLデータベースの libディレクトリを追記し、「OK」

を押します。（実際に DBをインストールしたディレクトリ名を入力してください） 

 

;C:¥Program Files¥MySQL¥MySQL Server 5.6¥lib¥ 

WPS ワークベンチを起動している状態の場合は、Path の設定後、WPS ワークベンチを再

起動6してください。 

 

ネイティブドライバーは、一般に ODBC ドライバーより高速に DB アクセスができる点で

メリットがあると言われています。 

WPS の MYSQL エンジンはネイティブドライバーを経由して MySQL データベースとアク

セスを行う仕組みを提供します。 

 

［SQL 言語を使用した DB 読み取り］ 

 

DB へのアクセスは SQL 言語を用いて行うことが多く、WPS でも SQL プロシジャを用い

て SQL言語による DBアクセスが可能です。 

 

SQL 言語を用いて WPS からネイティブドライバー経由で MySQL の DB アクセスを行う

場合は、以下のように、エンジン名として MYSQL を指定し、CONNECT 文のカッコ内に 

server=サーバー名、database=DB名、user=ユーザー名、password=パスワード などの接

続オプションや、後述する dbmax_text=n（文字テキストの読み取り最大長さの設定）など

の libnameオプションを指定します。 

 

以下は sakilaデータベースの cityテーブルから、country_id=50に該当するレコードのみ抽

出した全カラムを含むWPSデータセットWORK.japan_cityを作成しています。 

 

proc sql; 
  connect to MYSQL (server=localhost database=sakila user=XXX 
password=password); 
  create table japan_city as 
  select * 
  from connection to MYSQL 
  ( 
    select * 
    from city 
    where country_id=50 
  ); 
  disconnect from MYSQL; 
quit; 

※ 2行目の user=XXX password=password の XXXと passwordにはそれぞれユーザ名とパ

スワードを指定します。 

3行目の create table 文はWPSデータセット WORK.japan_cityを作成する指定です。 

                                                   
6 メニューバーの「ファイル(F)」 → 「再起動」をクリックします。 

8 行目の from 節で指定した city は sakila データベースの city テーブルを参照していま

す。 

 

(WPSワークベンチの実行ログ) 

8         proc sql; 
9           connect to MYSQL (server=localhost database=sakila user=XXX password=XXXX); 
NOTE: Successfully connected to database MYSQL as alias MYSQL. 
10          create table japan_city as 
11          select * 
12          from connection to MYSQL 
13          ( 
14            select * 
15            from city 
16            where country_id=50 
17          ); 
NOTE: Data set "WORK.japan_city" has 31 observation(s) and 4 variable(s) 
18          disconnect from MYSQL; 
NOTE: Successfully disconnected from database MYSQL. 
19        quit; 
NOTE: Procedure sql step took : 
      real time : 0.015 
      cpu time  : 0.000 

 

作成されたWPSデータセットWORK.staffデータセットの内容を確認します。 

 

※ 注意：MYSQL ドライバーによる DB へのアクセス制限等の設定によっては、SQL プロ

シジャの CONNECT 文の（）内のオプションに、その他の指定が必要となる場合がありま

す。指定可能なオプションは、以下のように、ヘルプビューで確認できます。 

まず、ヘルプビュー7の左上にある「内容」をクリックします。ヘルプ全体の目次が表示さ

れますので、WPS Reference for Language Elements をクリックします。 

 

WPSレファレンスを展開し、WPS Engine for MySQL を選択します。 

 

 

Library Engines → MySQLを選択すると、データベース接続オプション（connection option）

が表示されます。 

 

                                                   
7 ヘルプビューが存在しない場合は、「ウィンドウ(W)」→「ビューの表示(V)」→ 「その他(O)」→「ヘルプ」フォル

ダー → 「ヘルプ」からビューを出現させます。また、「ヘルプ(H)」→「ヘルプ目次(H)」から別ウィンドウで表示さ

れるメニューでも同じです。 

これらの接続オプション（user= password= database= server= port=）は MYSQLドライバ

ー経由でのデータベースアクセスの際の CONNECT 文の DBMS オプション指定としても、

また、次に節で説明する libnameステートメントによる DB接続オプションとしても用いる

ことができ、必要に応じて指定します。 

 

［libname ステートメントによる DB 入力］ 

 

WPSのエディタから以下のプログラムを入力し、実行します。 

 

libname in MYSQL database=sakila user=XXX password=password; 

※ user=XXX password=password の XXX と password にはそれぞれユーザ名とパスワー

ドを指定します。 

 

この指定は、WPS がデータベース sakila に MYSQL ネイティブドライバー経由で接続し、

sakila データベースのテーブルとビューをライブラリ名 in で参照できるようにする指定で

す。（入出力両方可能） ただし、DBへの接続やデータテーブルの検索に必要な権限の設定

や、他の接続オプションが必要となる場合があります。 

 

さらに、libname ステートメントには dbmax_text=n（文字テキストの読み取り最大長さの

設定）などの libname オプションを追加指定できます。以下のような libname オプション

が指定可能です。（前出の MYSQLエンジンのヘルプ画面から確認できます。） 

さて、実行すると、ログに 「ライブラリ名 in が割り当てられた」というメッセージが表

示されます。このメッセージは MYSQL エンジン経由で DB と接続できたことを表します。 

 

29        libname in MYSQL database=sakila user=XXX password=XXXX; 
NOTE: Library in assigned as follows: 
      Engine:        MYSQL 
      Physical Name:  

 

実行後、WPS サーバーエクスプローラービュー内の ローカルサーバー の下の Libraries 

を展開し、in ライブラリがあることを確認してください。この中に sakila データベースの

テーブル名およびビュー名が含まれていることがわかります。 

 



31 
 

 

※ 右矢印がついたアイテムはテーブル名（展開すると、テーブルに含まれる項目名と項目

タイプを表すアイコンが表示されます。）、付いていないのはビュー名です。 

 

これらのテーブルやビューは、WPSデータセットに対する操作と同じファイル操作が可能

です。例えば、以下のように、テーブル名を右クリックして出現するメニューの「プロパテ

ィ」を選択すると、カラム名などを確認することができます。 

 

 

 

 

 

 

※ 【重要な注意】：テーブル名をダブルクリックすると、データセットビューアが起動し、

テーブルの中身を表示しようとします。この操作はテーブルのサイズが大きいときはWPS

セッションが動かなくなったり不意に終了してしまうなどの危険を伴いますので、できる

だけ避けてください。 

 

さて、MYSQLエンジンを libnameステートメントで指定した後は、以下のプログラムのよ

うに、DB テーブルやビューをWPS データセットに変換することなく、直接プロシジャ入

力に用いることもできます。この例では、film テーブルを入力として映画鑑賞年齢制限

（rating）別の映画の本数の棒グラフを描いています。 

 

goptions cback=white; 
title "FILM Numbers by Rating"; 
proc gchart data=in.film; 
  vbar rating/patternid=midpoint freq width=20 space=5; 
run;quit;   

 

 

※ 注意：グラフィック表示はWPS3.1.1では日本語の表示はサポートされていません。 

 

また、以下のように、SAS 言語の DATA ステップでも DB テーブルを読むことができるよ

うになります。また、SQLプロシジャを使用して処理することも可能ですので、libnameス

テートメントを使って DBアクセスを行うときは、SAS言語と SQL言語いずれも使用可能

になるというメリットがあります。 

 

下の例は、film テーブルから題名（TITLE）に"BIRD"が含まれるレコードを選択した変数

File_id, title, rental_rate, rating を含むWPS データセットWORK.film_BIRD を DATA ステ

ップで作成しています。 

 

data film_BIRD; 



33 
 

  set in.film(keep=film_id title rental_rate rating); 
  where title contains "BIRD"; 
run; 

 

［WPS データセットの DB への出力方法］ 

 

以下のいずれかの方法でWPSデータセットを DBのテーブルへ出力できます。 

 

① DBLOADプロシジャ 

② SQLプロシジャ 

③ DATAステップ 

 

①の方法では DB への接続方法を DBLOAD プロシジャで指定しますので、libname ステー

トメントで DBエンジンを指定する必要はありません。しかし、②と③の方法では、事前に

DB への接続に用いるエンジン名と接続先（DB 名など）を libname ステートメントで指定

しておく必要があります。 

 

① DBLOADプロシジャ 

 

前節の最後のプログラムで filmテーブルから題名に"BIRD"を含むレコードを選択したWPS

データセットWORK.film_BIRDを MySQL の sakilaデータベースの film_BIRD テーブルへ

出力してみましょう。以下のようにプログラムを書いて実行します。 

 

proc dbload data=film_BIRD dbms=MYSQL; 
  server=localhost; 
  user=XXX; 
  password=password; 
  database=sakila; 
  table=film_BIRD; 
  load; 
run;quit; 

※ 注意： DBにアクセスするため MYSQLエンジンの接続オプションや libnameオプショ

ンがさらに必要な場合は、loadステートメントの前に、オプション名=値; の形式で指定を

追加してください。また、user=XXX password=password の XXXと passwordにはそれぞ

れユーザ名とパスワードを指定します。 

 

② SQLプロシジャによるテーブルへの読み書き 

 

Libnameステートメントでエンジン名を指定すると、DBテーブルをWPSデータセットと

同じように扱えるようになります。 

以下の例は sakilaデータベースの cityテーブルを読んで、country_id=50に該当するレコー

ドのみ抽出した全カラムを含むテーブル japan_cityを作成しています。 

 

libname in MYSQL database=sakila user=XXX password=password; 
proc sql; 
  create table in.japan_city as 
  select * 
  from in.city 
  where country_id=50; 
quit; 

※ この例では、in ライブラリは MySQL の sakila データベースを参照するよう最初の

libname ステートメントで定義しています。したがって、SQL プロシジャの中の in ライブ

ラリは WPSデータライブラリではなく、sakila DBを表すものと解釈され、読み取りも書

き出しも DBテーブルが対象になります。 

 

③ DATAステップ 

 

以下の例は②の SQL プロシジャを使った処理と同じことを SAS 言語の DATA ステッププ

ログラムで行っています。 

 

libname in MYSQL database=sakila user=XXX password=password; 

/* 同じ名前のテーブルを一旦削除 */ 
proc datasets lib=in; 
  delete japan_city; 
run;quit;   

/* DATAステップでDBテーブルを読み書き */ 
data in.japan_city; 
  set in.city; 
  where country_id=50; 
run;   

 

3.2. ODBC ドライバー経由の DB テーブルの入出力 

 

ODBC ドライバーは、Windows 環境や Linux 環境から各種 DB とのインタフェースを共通

方式で行うためのソフトウェアです。ほとんどの DB の ODBC ドライバーが DB ベンダや

サードパーティから提供されており、ODBCドライバーの設定を行えば、DBごとの細かい

アクセス条件等の設定を気にせずに、共通の構文で DBアクセスが実現できるという利点が

あります。 

［ODBC ドライバのインストール］ 

 

MySQL の ODBC ドライバーは Connector/ODBC を MySQL のバージョンに合わせてイン

ストールします。 もしも、次の［PC側の ODBCドライバーの設定］で MySQLの ODBC

ドライバーが見当たらない場合は、インストール環境を確認してください。 

 

［PC 側の ODBC ドライバーの設定］ 

 

WPSが導入されている PC 側では、以下のような手順で、MYSQL の sakila データベース

を ODBCデータソースとして登録しておきます。 

 

コントロールパネル → システムとセキュリティ → 管理ツールで ODBC データソースを

選択します。 

 

 

（左図）「追加」をクリックします。 

（右図）リストにある MySQL用 ODBCドライバーを選択し、「完了」を押します。 

 

 

 

（左図）新規データソースの名前と説明にはユーザーが自由に入力できます。ここでは、名

前に「MYSQL_SAKILA」、説明には「MySQL sakila database」と入力します。ユーザー名

とパスワードを入力すると、データベース欄のドロップリストが利用できますので、「sakila」

を選択します。次に「Test」を押して接続テストを行います。接続テストが成功することを

確認すると設定終了です。 

（右図）「OK」を押して ODBC データソースリストに MySQL 用 ODBC ドライバー

「MYSQL_SAKILA」が追加されたことを確認します。 

 

 

 

 

 

［SQL 言語を使用した DB 読み取り］ 

 

SQL 言語を用いてWPS から ODBCドライバー経由で MySQL の DBアクセスを行う場合

は、以下のように、エンジン名として ODBC を指定し、CONNECT 文のカッコ内に 

datasrc=ODBC データソース名などの接続オプションや、dbmax_text=n（文字テキストの

読み取り最大長さの設定）などの libnameオプションを指定します。 

 

下記は、ODBCドライバーを用いて、sakilaデータベースの cityテーブルから、country_id=50

に該当するレコードのみ抽出した全カラムを含む WPS データセット WORK.japan_city を

作成する例となっています。 

 

proc sql; 
  connect to ODBC (datasrc=MYSQL_SAKILA); 
  create table japan_city as 
  select * 
  from connection to ODBC 
  ( 
    select * 
    from city 
    where country_id=50 
  ); 
  disconnect from ODBC; 

quit; 

 

(WPSワークベンチの実行ログ) 

8         proc sql; 
9           connect to ODBC (datasrc=MYSQL_SAKILA); 
NOTE: Successfully connected to database ODBC as alias ODBC. 
10          create table japan_city as 
11          select * 
12          from connection to ODBC 
13          ( 
14            select * 
15            from city 
16            where country_id=50 
17          ); 
NOTE: Data set "WORK.japan_city" has 31 observation(s) and 4 variable(s) 
18          disconnect from ODBC; 
NOTE: Successfully disconnected from database ODBC. 
19        quit; 
NOTE: Procedure sql step took : 
      real time : 0.031 
      cpu time  : 0.015 

※ ODBCドライバーによる DBへのアクセス制限等の設定によっては、SQLプロシジャの

CONNECT文の（）内のオプションに、その他の指定が必要となる場合があります。以下の

ようにヘルプビューで指定可能なオプションが確認できます。 

 

ヘルプビュー8の左上にある「内容」をクリックします。 

ヘルプ全体の目次が表示されます。WPS Reference for Language Elements をクリックし

ます。 

 

 

WPSレファレンスを展開し、WPS Engine for ODBC を選択します。 

                                                   
8 ヘルプビューが存在しない場合は、「ウィンドウ(W)」→「ビューの表示(V)」→ 「その他(O)」→「ヘルプ」フォル

ダー → 「ヘルプ」からビューを出現させます。また、「ヘルプ(H)」→「ヘルプ目次(H)」から別ウィンドウで表示さ

れるメニューでも同じです。 

その後、展開を何度か行うと接続オプション（connection option） が表示されます。 

 

 

これらのオプション（user= password= datasrc= noprompt= prompt= readbuff= required=）

は ODBC ドライバー経由でのデータベースアクセスの際の CONNECT 文の DBMS オプシ

ョン指定としても、また、次に説明する libnameステートメントによるデータベースアクセ

スの接続オプションとしても用いることができ、必要に応じて指定します。 

 

［libname ステートメントによる DB 入力］ 

 

WPSのエディタから以下のプログラムを実行します。 

 

libname in ODBC datasrc=MYSQL_SAKILA; 

 

この指定は、WPSがデータソース MYSQL_SAKILA に ODBCドライバー経由で接続し、ラ

イブラリ名 inで参照できるようにする指定です。（入出力両方可能） 

ただし、DBへの接続やデータテーブルの検索に必要な権限の設定や、他の接続オプション

が必要となる場合があります。 

 

さらに、libname ステートメントには schema=スキーマ名、dbmax_text=n（文字テキスト

の読み取り最大長さの設定）などの libname オプションを追加指定できます。指定可能な

libnameオプションは MYSQLエンジンの libnameオプションとほぼ同じですが、詳細は前

出の ODBC エンジンのヘルプ画面から確認してください。（Connection-option の下に

libname-optionが記載されています。） 

 

実行すると、ログに「ライブラリ名 in が割り当てられた」というメッセージが表示されま

す。このメッセージは ODBC経由で DBと接続できたことを表します。 

 

345       libname in ODBC datasrc=MYSQL_SAKILA; 
NOTE: Library in assigned as follows: 
      Engine:        ODBC 
      Physical Name: MYSQL_SAKILA 

 

実行後、WPS サーバーエクスプローラービュー内の ローカルサーバー の下の Libraries 

を展開し、in ライブラリがあることを確認してください。この中に sakila データベースの

テーブル名およびビュー名が含まれていることがわかります。 

※ 右矢印がついたアイテムはテーブル名（展開すると、テーブルに含まれる項目名と項目

タイプを表すアイコンが表示されます。）、付いていないのはビュー名です。 

 

これらのテーブルやビューは、WPSデータセットに対する操作と同じファイル操作が可能

です。例えば、以下のように、テーブル名を右クリックして出現するメニューの「プロパテ

ィ」を選択すると、カラム名などを確認することができます。 

 

 

 

 

 

※ 【重要な注意】：テーブル名をダブルクリックすると、データセットビューアが起動し、

テーブルの中身を表示しようとします。この操作はテーブルのサイズが大きいときはWPS

セッションが動かなくなったり不意に終了してしまうなどの危険を伴いますので、できる

だけ避けてください。 

 

さて、ODBC エンジンを libname ステートメントで指定した後は、以下のプログラムのよ

うに、DB テーブルやビューをWPS データセットに変換することなく、直接プロシジャ入

力に用いることもできます。この例では、film テーブルを入力として映画鑑賞年齢制限

（rating）別の映画の本数の棒グラフを描いています。 

 

goptions cback=white; 
title "FILM Numbers by Rating"; 
proc gchart data=in.film; 
  vbar rating/patternid=midpoint freq width=20 space=5; 
run;quit;   

 

 

※ 注意：グラフィック表示はWPS3.1.1では日本語の表示はサポートされていません。 

 

また、以下のように、SAS 言語の DATA ステップでも DB テーブルを読むことができるよ

うになります。また、SQLプロシジャを使用して処理することも可能ですので、libnameス

テートメントを使って DBアクセスを行うときは、SAS言語と SQL言語いずれも使用可能

になるというメリットがあります。 

 

下の例は、film テーブルから題名（TITLE）に"BIRD"が含まれるレコードを選択した変数

File_id, title, rental_rate, rating を含むWPS データセットWORK.film_BIRD を DATA ステ

ップで作成しています。 

 

data film_BIRD; 
  set in.film(keep=film_id title rental_rate rating); 

  where title contains "BIRD"; 
run; 

 

［WPS データセットの DB への出力方法］ 

 

以下のいずれかの方法でWPSデータセットを DBのテーブルへ出力できます。 

 

① DBLOADプロシジャ 

② SQLプロシジャ 

③ DATAステップ 

 

①の方法では DB への接続方法を DBLOAD プロシジャで指定しますので、libname ステー

トメントで DBエンジンを指定する必要はありません。しかし、②と③の方法では、事前に

DB への接続に用いるエンジン名と接続先（DB 名やスキーマ名）を libname ステートメン

トで指定しておく必要があります。 

 

① DBLOADプロシジャ 

 

前節の最後のプログラムで sakiaテーブルから題名（TITLE）に"BIRD"が含まれるレコード

を選択した変数 File_id, title, rental_rate, ratingを含むWPSデータセットWORK.film_BIRD

を MySQLの sakilaデータベースの film_BIRD テーブルへ出力してみましょう。以下のよう

にプログラムを書いて実行します。 

 

proc dbload data=film_BIRD dbms=ODBC; 
  datasrc=MYSQL_SAKIA; 
  table=film_BIRD; 
  load; 
run;quit; 

※ 注意： DBにアクセスするための権限や ODBCエンジンの接続オプションや libnameオ

プションがさらに必要な場合は、loadステートメントの前に、オプション名=値; の形式で

指定を追加してください。 

 

② SQLプロシジャによるテーブルへの読み書き 

 

Libnameステートメントでエンジン名を指定すると、DBテーブルをWPSデータセットと

同じように扱えるようになります。 

以下の例は sakilaデータベースの cityテーブルを読んで、country_id=50に該当するレコー

ドのみ抽出した全カラムを含むテーブル japan_cityを作成しています。 

libname in ODBC datasrc=MYSQL_SAKILA; 
proc sql; 
  create table in.japan_city as 
  select * 
  from in.city 
  where country_id = 50; 
quit; 

※ この例では、in ライブラリは ODBC 経由で sakila データベースを参照するよう最初の

libname ステートメントで定義しています。したがって、SQL プロシジャの中の in ライブ

ラリは WPSデータライブラリではなく、sakila DBを表すものと解釈され、読み取りも書

き出しも DBテーブルが対象になります。 

 

③ DATAステップ 

 

以下の例は②の SQL プロシジャを使った処理と同じことを SAS 言語の DATA ステッププ

ログラムで行っています。 

 

libname in ODBC datasrc=MYSQL_SAKILA; 

/* 同じ名前のテーブルを一旦削除 */ 
proc datasets lib=in; 
  delete japan_city; 
run;quit;   

/* DATAステップでDBテーブルを読み書き */ 
data in.japan_city; 
  set in.city; 
  where country_id=50; 
run;   

 

 

4. ORACLE とのインタフェース 

 

［scott サンプルスキーマ］ 

 

$ORACLE_HOME/rdbms/admin/utlsampl.sql を sqlplus から実行して作成される scott ス

キーマ上の DEPTテーブルと EMPテーブルを例として用います。 

4.1. ネイティブドライバー経由の DB テーブルの入出力 

 

まず、コマンドラインインタフェース（CLIドライバー、ネイティブドライバー）を用いて

WPSから DBテーブルの入出力を行ってみましょう。 

ネイティブドライバーは、DB と一緒にインストールされます。一般に ODBC ドライバー

より高速に DBアクセスができる点でメリットがあると言われています。 

WPSのORACLEエンジンはネイティブドライバーを経由してORACLEデータベースとア

クセスを行う仕組みを提供します。 

 

［SQL 言語を使用した DB 読み取り］ 

 

DB へのアクセスは SQL 言語を用いて行うことが多く、WPS でも SQL プロシジャを用い

て SQL言語による DBアクセスが可能です。 

 

SQL言語を用いてWPSからネイティブドライバー経由でORACLE のDBアクセスを行う

場合は、以下のように、エンジン名として ORACLE を指定し、CONNECT文のカッコ内に 

user=ユーザー名、password=パスワード、path=パス の接続オプションや schema=、

dbmax_text=n（文字テキストの読み取り最大長さの設定）などの libname オプションを指

定します。 

 

以下は scottスキーマ上の EMPテーブルから、SAL>2000に該当するレコードのみ抽出し

た全カラムを含むWPSデータセットWORK.EMP_SAL_OVER2000を作成しています。 

proc sql; 
  connect to ORACLE (user=XXX password=password); 
  create table EMP_SAL_OVER2000 as 
  select * 
  from connection to ORACLE 
  ( 
    select * 
    from scott.EMP 
    where SAL>2000 
   ); 
  disconnect from ORACLE; 
quit;    

※ 2行目の user=XXX password=password の XXXと passwordにはそれぞれユーザ名とパ

スワードを指定します。 

3 行目の create table 文は WPS データセット WORK.EMP_SAL_OVER2000 を作成する

指定です。 

8行目の from節で指定した scott.EMP は scottスキーマ上の EMPテーブルを参照してい

ます。 

 

(WPSワークベンチの実行ログ) 

8         proc sql; 
9           connect to ORACLE (user=XXX password=XXXXXXXX); 
NOTE: Successfully connected to database ORACLE as alias ORACLE. 
10          create table EMP_SAL_OVER2000 as 
11          select * 
12          from connection to ORACLE 
13          ( 
14            select * 
15            from scott.EMP 
16            where SAL>2000 
17           ); 
NOTE: Data set "WORK.EMP_SAL_OVER2000" has 5 observation(s) and 8 variable(s) 
18          disconnect from ORACLE; 
NOTE: Successfully disconnected from database ORACLE. 
19        quit; 
NOTE: Procedure sql step took : 
      real time : 0.078 
      cpu time  : 0.031 

 

作成された WPS データセット WORK. EMP_SAL_OVER2000 データセットの内容を確認

します。 

 

 

※ 注意：ORACLE ドライバーによる DB へのアクセス制限等の設定によっては、SQL プ

ロシジャの CONNECT 文の（）内のオプションに、その他の指定が必要となる場合があり

ます。指定可能なオプションは、以下のように、ヘルプビューで確認できます。 

 

まず、ヘルプビュー9の左上にある「内容」をクリックします。ヘルプ全体の目次が表示さ

れますので、WPS Reference for Language Elements をクリックします。 

 

WPSレファレンスを展開し、WPS Engine for ORACLE を選択します。 

 

 

Library Engines → ORACLE を選択すると、データベース接続オプション（connection 

option）が表示されます。 

 

                                                   
9 ヘルプビューが存在しない場合は、「ウィンドウ(W)」→「ビューの表示(V)」→ 「その他(O)」→「ヘルプ」フォル

ダー → 「ヘルプ」からビューを出現させます。また、「ヘルプ(H)」→「ヘルプ目次(H)」から別ウィンドウで表示さ

れるメニューでも同じです。 

これらの接続オプション（user= password= path=）は ORACLE ドライバー経由でのデー

タベースアクセスの際の CONNECT文の DBMSオプション指定としても、また、次に節で

説明する libnameステートメントによる DB接続オプションとしても用いることができ、必

要に応じて指定します。 

 

［libname ステートメントによる DB 入力］ 

 

WPSのエディタから以下のプログラムを入力し、実行します。 

 

libname in ORACLE user=XXX password=password schema=scott; 

※ user=XXX password=password の XXX と password にはそれぞれユーザ名とパスワー

ドを指定します。 

 

この指定は、WPSがスキーマ scottに ORACLEネイティブドライバー経由で接続し、scott

スキーマ上のテーブルとビューをライブラリ名 inで参照できるようにする指定です。（入出

力両方可能） ただし、DBへの接続やデータテーブルの検索に必要な権限の設定や、他の接

続オプションが必要となる場合があります。 

 

さらに、libname ステートメントには shema=スキーマ名、 dbmax_text=n（文字テキスト

の読み取り最大長さの設定）などの libnameオプションを追加指定できます。以下のような

libnameオプションが指定可能です。（前出の ORACLE エンジンのヘルプ画面から確認でき

ます。） 

さて、実行すると、ログに 「ライブラリ名 in が割り当てられた」というメッセージが表

示されます。このメッセージは ORACLEエンジン経由で DBと接続できたことを表します。 

 

49        libname in ORACLE user=XXX password=XXXXXX schema=scott; 
NOTE: Library in assigned as follows: 
      Engine:        ORACLE 
      Physical Name: 

 

実行後、WPS サーバーエクスプローラービュー内の ローカルサーバー の下の Libraries 

を展開し、inライブラリがあることを確認してください。この中に scottスキーマ上のテー

ブル名が含まれていることがわかります。 

※ 右の白抜き矢印がついたアイテムはテーブル名（展開すると、テーブルに含まれる項目

名と項目タイプを表すアイコンが表示されます。）、この scottスキーマにはビューが存在し

ませんが、ビュー名の場合には矢印が付きません。 

 

これらのテーブルやビューは、WPSデータセットに対する操作と同じファイル操作が可能

です。例えば、以下のように、テーブル名を右クリックして出現するメニューの「プロパテ

ィ」を選択すると、カラム名などを確認することができます。 

 

 

 

 

 

※ 【重要な注意】：テーブル名をダブルクリックすると、データセットビューアが起動し、

テーブルの中身を表示しようとします。この操作はテーブルのサイズが大きいときはWPS

セッションが動かなくなったり不意に終了してしまうなどの危険を伴いますので、できる

だけ避けてください。 

 

さて、ORACLE エンジンを libname ステートメントで指定した後は、以下のプログラムの

ように、DB テーブルやビューをWPS データセットに変換することなく、直接プロシジャ

入力に用いることもできます。この例では、EMP テーブルを入力として職種（JOB）別給

与（SAL）平均を示す棒グラフを描いています。 

 

goptions cback=white; 
title "Average Salary by Job"; 

proc gchart data=in.emp; 
  vbar job/sumvar=SAL type=mean patternid=midpoint mean width=20 space=5; 
run;quit;   

 

 

※ 注意：グラフィック表示はWPS3.1.1では日本語の表示はサポートされていません。 

 

また、以下のように、SAS 言語の DATA ステップでも DB テーブルを読むことができるよ

うになります。また、SQLプロシジャを使用して処理することも可能ですので、libnameス

テートメントを使って DBアクセスを行うときは、SAS言語と SQL言語いずれも使用可能

になるというメリットがあります。 

 

下の例は、EMPテーブルの中で直属の上司（MGR）が社員名（ENAME）"BLAKE" の社員

番号（EMPNO）を持つ社員のみを抽出したWPSデータセットWORK.BLAKE_grpを DATA

ステップで作成しています。 

 

data BLAKE_grp; 
  if _n_=1 then set in.emp(keep=ENAME EMPNO where=(ENAME="BLAKE") 
rename=(EMPNO=SEARCHNO)); 
  set in.emp; 
  if MGR=SEARCHNO; 
run; 

 

［WPS データセットの DB への出力方法］ 

 

以下のいずれかの方法でWPSデータセットを DBのテーブルへ出力できます。 

 

① DBLOADプロシジャ 

② SQLプロシジャ 

③ DATAステップ 

 

①の方法では DB への接続方法を DBLOAD プロシジャで指定しますので、libname ステー

トメントで DBエンジンを指定する必要はありません。しかし、②と③の方法では、事前に

DB への接続に用いるエンジン名と接続先（DB 名など）を libname ステートメントで指定

しておく必要があります。 

 

① DBLOADプロシジャ 

 

前節の最後のプログラムでEMPテーブルから"BLAKE"を直属の上司とする社員を選択した

WPSデータセットWORK.BLAKE_grpを ORACLE の scottスキーマ上の BLAKE_GRPテ

ーブルへ出力してみましょう。以下のようにプログラムを書いて実行します。 

 

proc dbload data=BLAKE_grp dbms=ORACLE; 
  user=XXX; 
  password=password; 
  schema=scott; 
  table=BLAKE_GRP; 
  load; 
run;quit; 

※ 注意： DB にアクセスするため ORACLE エンジンの接続オプションや libname オプシ

ョンがさらに必要な場合は、loadステートメントの前に、オプション名=値; の形式で指定

を追加してください。また、user=XXX password=password の XXXと passwordにはそれ

ぞれユーザ名とパスワードを指定します。 

 

② SQLプロシジャによるテーブルへの読み書き 

 

Libnameステートメントでエンジン名を指定すると、DBテーブルをWPSデータセットと

同じように扱えるようになります。 

以下の例は scott スキーマ上の dept テーブルを読んで、LOC="CHICAGO"に該当するレコ

ードのみ抽出した全カラムを含む 1 件のレコードのみ含むテーブル CHICAGO を作成して

います。 

 

libname in ORACLE user=XXX password=password schema=scott; 
proc sql; 
  create table in.CHICAGO as 
  select * 
  from in.dept 
  where LOC="CHICAGO"; 
quit; 

※ この例では、inライブラリは ORACLEデータベースの scottスキーマを参照するよう最

初の libnameステートメントで定義しています。したがって、SQLプロシジャの中の inラ

イブラリは WPS データライブラリではなく、scott スキーマを表すものと解釈され、読み

取りも書き出しも DBテーブルが対象になります。 

 

③ DATAステップ 

 

以下の例は②の SQL プロシジャを使った処理と同じことを SAS 言語の DATA ステッププ

ログラムで行っています。 

 

libname in ORACLE user=XXX password=password schema=scott; 

/* 同じ名前のテーブルを一旦削除 */ 
proc datasets lib=in; 
  delete CHICAGO; 
run;quit;   

/* DATAステップでDBテーブルを読み書き */ 
data in.CHICAGO; 
  set in.DEPT; 
  where LOC="CHICAGO"; 
run;   

 

4.2. ODBC ドライバー経由の DB テーブルの入出力 

 

ODBC ドライバーは、Windows 環境や Linux 環境から各種 DB とのインタフェースを共通

方式で行うためのソフトウェアです。ほとんどの DB の ODBC ドライバーが DB ベンダや

サードパーティから提供されており、ODBCドライバーの設定を行えば、DBごとの細かい

アクセス条件等の設定を気にせずに、共通の構文で DBアクセスが実現できるという利点が

あります。 

 

［ODBC ドライバのインストール］ 

 

［PC 側の ODBC ドライバーの設定］ 

 

WPSが導入されている PC側では、以下のような手順で、ORACLE データベースを ODBC

データソースとして登録しておきます。 

 

コントロールパネル → システムとセキュリティ → 管理ツールで ODBC データソースを

選択します。 

（左図）「追加」をクリックします。 

（右図）リストにある ORACLE用 ODBCドライバーを選択し、「完了」を押します。 

 

 

（左図）新規データソースの名前と説明にはユーザーが自由に入力できます。ここでは、

Data Source Nameに「XE」、Descriptionには「ORACLE XE ODBC Driver」と入力します。 

「Test Connection」ボタンを押して User Nameと Passwordを入力し、「OK」ボタンを押

して「Connection Successful」のメッセージが出て接続テストが成功することを確認する

と設定終了です。 

（右図）「OK」を押して ODBC データソースリストに ORACLE 用 ODBC データソース

「XE」が追加されたことを確認します。 

［SQL 言語を使用した DB 読み取り］ 

 

SQL 言語を用いて WPS から ODBC ドライバー経由で ORACLE の DB アクセスを行う場

合は、以下のように、エンジン名として ODBCを指定し、CONNECT文のカッコ内に user=

ユーザー名、password=パスワード、datasrc=ODBCデータソース名などの接続オプション

や dbmax_text=n（文字テキストの読み取り最大長さの設定）などの libnameオプションを

指定します。 

 

下記は、ODBCドライバーを用いて、scottスキーマ上の EMPテーブルから、SAL>2000に

該 当 す る レ コ ー ド の み 抽 出 し た 全 カ ラ ム を 含 む WPS デ ー タ セ ッ ト

WORK.EMP_SAL_OVER2000を作成しています。 

 

proc sql; 
  connect to ODBC (user=XXX password=password datasrc=XE); 
  create table EMP_SAL_OVER2000 as 
  select * 
  from connection to ODBC 
  ( 
    select * 
    from scott.EMP 
    where SAL>2000 
   ); 
  disconnect from ODBC; 
quit;    

※ 2行目の user=XXX password=password の XXXと passwordにはそれぞれユーザ名とパ

スワードを指定します。 

 

(WPSワークベンチの実行ログ) 

146       proc sql; 
147         connect to ODBC (user=XXX password=XXXXXX datasrc=XE); 
NOTE: Successfully connected to database ODBC as alias ODBC. 
148         create table EMP_SAL_OVER2000 as 

149         select * 
150         from connection to ODBC 
151         ( 
152           select * 
153           from scott.EMP 
154           where SAL>2000 
155          ); 
NOTE: Data set "WORK.EMP_SAL_OVER2000" has 5 observation(s) and 8 variable(s) 
156         disconnect from ODBC; 
NOTE: Successfully disconnected from database ODBC. 
157       quit; 
NOTE: Procedure sql step took : 
      real time : 0.015 
      cpu time  : 0.000 

※ ODBCドライバーによる DBへのアクセス制限等の設定によっては、SQLプロシジャの

CONNECT文の（）内のオプションに、その他の指定が必要となる場合があります。以下の

ようにヘルプビューで指定可能なオプションが確認できます。 

 

ヘルプビュー10の左上にある「内容」をクリックします。 

ヘルプ全体の目次が表示されます。WPS Reference for Language Elements をクリックし

ます。 

 

 

WPSレファレンスを展開し、WPS Engine for ODBC を選択します。 

                                                   
10 ヘルプビューが存在しない場合は、「ウィンドウ(W)」→「ビューの表示(V)」→ 「その他(O)」→「ヘルプ」フォル

ダー → 「ヘルプ」からビューを出現させます。また、「ヘルプ(H)」→「ヘルプ目次(H)」から別ウィンドウで表示さ

れるメニューでも同じです。 

その後、展開を何度か行うと接続オプション（connection option） が表示されます。 

 

 

これらのオプション（user= password= datasrc= noprompt= prompt= readbuff= required=）

は ODBC ドライバー経由でのデータベースアクセスの際の CONNECT 文の DBMS オプシ

ョン指定としても、また、次に説明する libnameステートメントによるデータベースアクセ

スの接続オプションとしても用いることができ、必要に応じて指定します。 

 

［libname ステートメントによる DB 入力］ 

 

WPSのエディタから以下のプログラムを実行します。 

 

libname in ODBC datasrc=XE user=XXX password=password schema=scott; 

※ user=XXX password=password の XXXと passwordにはそれぞれユーザ名とパスワード

を指定します。 

 

この指定は、WPS がデータソース XE の scott スキーマ上のテーブルに ODBC ドライバー

経由で接続し、ライブラリ名 inで参照できるようにする指定です。（入出力両方可能） 

ただし、DBへの接続やデータテーブルの検索に必要な権限の設定や、他の接続オプション

（path=）が必要となる場合があります。 

 

さらに、libname ステートメントには schema=スキーマ名、dbmax_text=n（文字テキスト

の読み取り最大長さの設定）などの libname オプションを追加指定できます。指定可能な

libname オプションは ORACLE エンジンの libname オプションとほぼ同じですが、詳細は

前出の ODBC エンジンのヘルプ画面から確認してください。（Connection-option の下に

libname-optionが記載されています。） 

 

実行すると、ログに「ライブラリ名 in が割り当てられた」というメッセージが表示されま

す。このメッセージは ODBC経由で DBと接続できたことを表します。 

 

167       libname in ODBC datasrc=XE user=XXX password=XXXXXX schema=scott; 
NOTE: Library in assigned as follows: 
      Engine:        ODBC 
      Physical Name: XE 

 

実行後、WPS サーバーエクスプローラービュー内の ローカルサーバー の下の Libraries 

を展開し、inライブラリがあることを確認してください。この中に scottスキーマ上のテー

ブル名が含まれていることがわかります。 

※ 右の白抜き矢印がついたアイテムはテーブル名（展開すると、テーブルに含まれる項目

名と項目タイプを表すアイコンが表示されます。）、ビュー名の場合には矢印が付いていま

せん。 

 

これらのテーブルやビューは、WPSデータセットに対する操作と同じファイル操作が可能

です。例えば、以下のように、テーブル名を右クリックして出現するメニューの「プロパテ

ィ」を選択すると、カラム名などを確認することができます。 

 

 

 

 

 

※ 【重要な注意】：テーブル名をダブルクリックすると、データセットビューアが起動し、

テーブルの中身を表示しようとします。この操作はテーブルのサイズが大きいときはWPS

セッションが動かなくなったり不意に終了してしまうなどの危険を伴いますので、できる

だけ避けてください。 

 

さて、ODBC エンジンを libname ステートメントで指定した後は、以下のプログラムのよ

うに、DB テーブルやビューをWPS データセットに変換することなく、直接プロシジャ入

力に用いることもできます。この例では、EMP テーブルを入力として職種（JOB）別給与

（SAL）平均を示す棒グラフを描いています。 

 

goptions cback=white; 
title "Average Salary by Job"; 

proc gchart data=in.emp; 
  vbar job/sumvar=SAL type=mean patternid=midpoint mean width=20 space=5; 
run;quit;   

 

 

※ 注意：グラフィック表示はWPS3.1.1では日本語の表示はサポートされていません。 

 

また、以下のように、SAS 言語の DATA ステップでも DB テーブルを読むことができるよ

うになります。また、SQLプロシジャを使用して処理することも可能ですので、libnameス

テートメントを使って DBアクセスを行うときは、SAS言語と SQL言語いずれも使用可能

になるというメリットがあります。 

 

下の例は、EMPテーブルの中で直属の上司（MGR）が社員名（ENAME）"BLAKE" の社員

番号（EMPNO）を持つ社員のみを抽出したWPSデータセットWORK.BLAKE_grpを DATA

ステップで作成しています。 

 

data BLAKE_grp; 
  if _n_=1 then set in.emp(keep=ENAME EMPNO where=(ENAME="BLAKE") 
rename=(EMPNO=SEARCHNO)); 
  set in.emp; 
  if MGR=SEARCHNO; 
run; 

 

［WPS データセットの DB への出力方法］ 

 

以下のいずれかの方法でWPSデータセットを DBのテーブルへ出力できます。 

 

① DBLOADプロシジャ 

② SQLプロシジャ 

③ DATAステップ 

 

①の方法では DB への接続方法を DBLOAD プロシジャで指定しますので、libname ステー

トメントで DBエンジンを指定する必要はありません。しかし、②と③の方法では、事前に

DB への接続に用いるエンジン名と接続先（DB 名やスキーマ名）を libname ステートメン

トで指定しておく必要があります。 

 

① DBLOADプロシジャ 

 

前節の最後のプログラムで EMP テーブルの中で直属の上司（MGR）が社員名（ENAME）

"BLAKE" の社員番号（EMPNO）を持つ社員のみを抽出した WPS データセット

WORK.BLAKE_grp を ODBC ドライバー経由で XE データソースの scott スキーマ上の

BLAKE_GRP テーブルへ出力してみましょう。以下のようにプログラムを書いて実行しま

す。 

 

proc dbload data=BLAKE_grp dbms=ODBC; 
  datasrc=XE; 
  user=XXX; 
  password=password; 
  schema=scott; 
  table=BLAKE_GRP; 
  load; 
run;quit; 

※ 注意： user=XXX password=password の XXXと passwordにはそれぞれユーザ名とパス

ワードを指定します。また、DB にアクセスするための権限や ODBC エンジンの接続オプ

ションや libname オプションがさらに必要な場合は、load ステートメントの前に、オプシ

ョン名=値; の形式で指定を追加してください。 

 

 

② SQLプロシジャによるテーブルへの読み書き 

 

Libnameステートメントでエンジン名を指定すると、DBテーブルをWPSデータセットと

同じように扱えるようになります。 

以下の例は scott スキーマ上の dept テーブルを読んで、LOC="CHICAGO"に該当するレコ

ードのみ抽出した全カラムを含む 1 件のレコードのみ含むテーブル CHICAGO を作成して

います。 

 

libname in ODBC datasrc=XE user=XXX password=password schema=scott; 
proc sql; 

  create table in.CHICAGO as 
  select * 
  from in.dept 
  where LOC="CHICAGO"; 
quit; 

※ この例では、in ライブラリは ODBC 経由で ORACLE データベースの scott スキーマを

参照するよう最初の libname ステートメントで定義しています。したがって、SQL プロシ

ジャの中の inライブラリは WPSデータライブラリではなく、ORACLE DBを表すものと

解釈され、読み取りも書き出しも DBテーブルが対象になります。 

 

③ DATAステップ 

 

以下の例は②の SQL プロシジャを使った処理と同じことを SAS 言語の DATA ステッププ

ログラムで行っています。 

 

libname in ODBC datasrc=XE user=XXX password=password schema=scott; 

/* 同じ名前のテーブルを一旦削除 */ 
proc datasets lib=in; 
  delete CHICAGO; 
run;quit;   

/* DATAステップでDBテーブルを読み書き */ 
data in.CHICAGO; 
  set in.DEPT; 
  where LOC="CHICAGO"; 
run;   

 

 

5. PostregSQL とのインタフェース 

 

［dvdrental サンプル DB］ 

 

dvdrentalデータベース11を例として用います。 

 

                                                   
11 http://www.postgresqltutorial.com/load-postgresql-sample-database/ からダウンロードできます。 

http://www.postgresqltutorial.com/load-postgresql-sample-database/

5.1. ネイティブドライバー経由の DB テーブルの入出力 

 

まず、コマンドラインインタフェース（CLIドライバー、ネイティブドライバー）を用いて

WPSから DBテーブルの入出力を行ってみましょう。 

 

ネイティブドライバーは、一般に ODBC ドライバーより高速に DB アクセスができる点で

メリットがあると言われています。 

WPSの POSTGSQLエンジンはネイティブドライバーを経由して PostgreSQLデータベー

スとアクセスを行う仕組みを提供します。 

 

【重要な注意】PostgreSQLのネイティブドライバー経由のインタフェースの記述部分につ

いては、WPS3.1のビルド 678 (03.01.01.00.000678) 以降の対応状況を反映しています。 

 

［SQL 言語を使用した DB 読み取り］ 

 

DB へのアクセスは SQL 言語を用いて行うことが多く、WPS でも SQL プロシジャを用い

て SQL言語による DBアクセスが可能です。 

 

SQL 言語を用いて WPS からネイティブドライバー経由で PostgreSQL の DB アクセスを

行う場合は、以下のように、エンジン名として POSTGSQL を指定し、CONNECT 文のカ

ッコ内に database=データベース名、user=ユーザー名、password=パスワード などの接続

オプションや shema=スキーマ名、dbmax_text=n（文字テキストの読み取り最大長さの設

定）などの後述する libnameオプションを必要に応じて指定します。 

 

以下は dvdrentalデータベースの publicスキーマ上の cityテーブルから、country_id=50に

該当するレコードのみ抽出した全カラムを含む WPS データセット WORK.japan_city を作

成しています。 

 

proc sql; 
  connect to POSTGSQL (database=dvdrental user=XXX password=password); 
  create table japan_city as 
  select * 
  from connection to POSTGSQL 
  ( 
    select * 
    from public.city 
    where country_id=50 
  ); 
  disconnect from POSTGSQL; 
quit;    

※ 2行目の user=XXX password=password の XXXと passwordにはそれぞれユーザ名とパ

スワードを指定します。 

3行目の create table 文はWPSデータセット WORK.japan_cityを作成する指定です。 

8 行目の from 節で指定した public.city は dvdrental データベースの public スキーマ上の

cityテーブルを参照しています。 

 

(WPSワークベンチの実行ログ) 

8         proc sql; 
9           connect to POSTGSQL (database=dvdrental user=XXX password=XXXXXXX); 
NOTE: Successfully connected to database POSTGSQL as alias POSTGSQL. 
10          create table japan_city as 
11          select * 
12          from connection to POSTGSQL 
13          ( 
14            select * 
15            from public.city 
16            where country_id=50 
17          ); 
NOTE: Data set "WORK.japan_city" has 31 observation(s) and 4 variable(s) 
18          disconnect from POSTGSQL; 
NOTE: Successfully disconnected from database POSTGSQL. 
19        quit; 
NOTE: Procedure sql step took : 
      real time : 0.0093 
      cpu time  : 0.000 

 

作成されたWPSデータセットWORK.staffデータセットの内容を確認します。 

※ 【重要な注意】WPS3.1.1 の POSTGSQL エンジンでは WPS データセットのデータ型

に変換できないデータベースカラムのデータ型（ENUM型,mpaa_rating型, text[]型, tsvector

型）があり、これらのデータ型のカラムは除いて読み取る必要があります。対応としては後

述の ODBCドライバーを用いる方法もあります。 

 

※ 注意：POSTGSQL ドライバーによる DB へのアクセス制限等の設定によっては、SQL

プロシジャの CONNECT 文の（）内のオプションに、その他の指定が必要となる場合があ

ります。指定可能なオプションは、以下のように、ヘルプビューで確認できます。 

 

 

まず、ヘルプビュー12の左上にある「内容」をクリックします。ヘルプ全体の目次が表示さ

れますので、WPS Reference for Language Elements をクリックします。 

 

WPSレファレンスを展開し、WPS Engine for PostgreSQL を選択します。 

                                                   
12 ヘルプビューが存在しない場合は、「ウィンドウ(W)」→「ビューの表示(V)」→ 「その他(O)」→「ヘルプ」フォル

ダー → 「ヘルプ」からビューを出現させます。また、「ヘルプ(H)」→「ヘルプ目次(H)」から別ウィンドウで表示さ

れるメニューでも同じです。 

Library Engines →  POSTGRESQL を選択すると、データベース接続オプション

（connection option）が表示されます。 

 

 

これらの接続オプション（user= password= database= server= port=）は POSTGSQLドラ

イバー経由でのデータベースアクセスの際の CONNECT 文の DBMS オプション指定とし

ても、また、次に節で説明する libnameステートメントによる DB接続オプションとしても

用いることができ、必要に応じて指定します。 

 

［libname ステートメントによる DB 入力］ 

 

WPSのエディタから以下のプログラムを入力し、実行します。 

libname in POSTGSQL database=dvdrental user=XXX password=password 
schema=public; 

※ user=XXX password=password の XXX と password にはそれぞれユーザ名とパスワー

ドを指定します。 

 

この指定は、WPS がデータベース dvdrental の public スキーマに POSTGSQL ネイティブ

ドライバー経由で接続し、dvdrentalデータベースの publicスキーマ上のテーブルとビュー

をライブラリ名 inで参照できるようにする指定です。（入出力両方可能） ただし、DBへの

接続やデータテーブルの検索に必要な権限の設定や、他の接続オプションが必要となる場

合があります。 

 

さらに、libname ステートメントには dbmax_text=n（文字テキストの読み取り最大長さの

設定）などの libname オプションを追加指定できます。以下のような libname オプション

が指定可能です。（前出の POSTGSQLエンジンのヘルプ画面から確認できます。） 

 

 

 

さて、実行すると、ログに 「ライブラリ名 in が割り当てられた」というメッセージが表

示されます。このメッセージは POSTGSQLエンジン経由で DBと接続できたことを表しま

す。 

 

29        libname in POSTGSQL database=dvdrental user=XXX password=XXXXXXX schema=public; 
NOTE: Library in assigned as follows: 
      Engine:        POSTGRESQL 
      Physical Name: 

 

実行後、WPS サーバーエクスプローラービュー内の ローカルサーバー の下の Libraries 

を展開し、inライブラリがあることを確認してください。この中に dvdrentalデータベース

の publicスキーマ上のテーブル名およびビュー名が含まれていることがわかります。 

 

 

※ 右矢印がついたアイテムはテーブル名（展開すると、テーブルに含まれる項目名と項目

タイプを表すアイコンが表示されます。）、付いていないのはビュー名です。 

 

これらのテーブルやビューは、WPSデータセットに対する操作と同じファイル操作が可能

です。例えば、以下のように、テーブル名を右クリックして出現するメニューの「プロパテ

ィ」を選択すると、カラム名などを確認することができます。 

※ 【重要な注意 1】WPS3.1.1の POSTGSQLエンジンではWPSデータセットのデータ型

に変換できないデータ型（ENUM型,mpaa_rating型, text[]型, tsvector型）があり、これら

のデータ型のカラムを含むテーブルやビューは、アクセスエラーが発生し読み取りはでき

ません。対応としては後述の ODBCドライバーを用いてください。 

※ 【重要な注意 2】：テーブル名をダブルクリックすると、データセットビューアが起動し、

テーブルの中身を表示しようとします。この操作はテーブルのサイズが大きいときはWPS

セッションが動かなくなったり不意に終了してしまうなどの危険を伴いますので、できる

だけ避けてください。 

 

さて、POSTGSQLエンジンを libnameステートメントで指定した後は、以下のプログラム

のように、DB テーブルやビューをWPS データセットに変換することなく、直接プロシジ

ャ入力に用いることもできます。この例では、customer_listビューを入力として国（country）

別の登録顧客数の棒グラフを描いています。 

 

goptions cback=white; 
title "Customer number by Country"; 
proc gchart data=in.customer_list; 
  pie country; 
run;quit;   

※ 注意：グラフィック表示はWPS3.1.1では日本語の表示はサポートされていません。 

 

また、以下のように、SAS 言語の DATA ステップでも DB テーブルを読むことができるよ

うになります。また、SQLプロシジャを使用して処理することも可能ですので、libnameス

テートメントを使って DBアクセスを行うときは、SAS言語と SQL言語いずれも使用可能

になるというメリットがあります。 

 

下の例は、customer_list ビューから国名（country）を大文字化した値が"GREECE"のレコ

ードを選択した変数 id, name, address, city, country を含む WPS データセット

WORK.custromer_greeceを DATAステップで作成しています。 

 

data customer_greece; 
  set in.customer_list(keep=id name address city country); 
  where upcase(country)="GREECE"; 
run; 

 

［WPS データセットの DB への出力方法］ 

 

以下のいずれかの方法でWPSデータセットを DBのテーブルへ出力できます。 

 

① DBLOADプロシジャ 

 

前節の最後のプログラムで customer_list ビューから国名（country）を大文字化した値が

"GREECE"のレコードを選択した変数 id, name, address, city, countryを含むWPSデータ

セット WORK.custromer_greece を POSTGRESQL の dvdrental データベースの public ス

キームの customer_greece テーブルへ出力してみましょう。以下のようにプログラムを書

いて実行します。 

 

proc dbload data=customer_greece dbms=POSTGSQL; 
  user=XXX; 
  password=password; 
  database=dvdrental; 
  host=localhost;    
  port=5432;     
  table=public.customer_greece; 
  load; 
run;quit; 

※ 注意： DBにアクセスするため POSTGSQLエンジンの接続オプションや libnameオプ

ションがさらに必要な場合は、loadステートメントの前に、オプション名=値; の形式で指

定を追加してください。また、user=XXX password=password の XXXと passwordにはそ

れぞれユーザ名とパスワードを指定します。 

 

② SQLプロシジャによるテーブルへの読み書き 

 

Libnameステートメントでエンジン名を指定すると、DBテーブルをWPSデータセットと

同じように扱えるようになります。 

以下の例は dvdrental データベースの public スキーム上の city テーブルを読んで、

country_id=50に該当するレコードのみ抽出した全カラムを含むテーブル japan_cityを作成

しています。 

 

libname in POSTGSQL database=dvdrental user=XXX password=password 
schema=public; 
proc sql; 
  create table in.japan_city as 
  select * 
  from in.city 
  where country_id=50;  
quit; 

※ この例では、in ライブラリは dvdrental データベースの public スキームを参照するよう

最初の libname ステートメントで定義しています。したがって、SQL プロシジャの中の in

ライブラリは WPSデータライブラリではなく、dvdrentalデータベースの publicスキーム

を表すものと解釈され、読み取りも書き出しも DBテーブルが対象になります。 

【重要な注意】現時点では、実行後、再度 libnameステートメントを実行してください。そ

うでないと、データセットの内容やプロパティを確認できないという問題があります。 

 

③ DATAステップ 

以下の例は②の SQL プロシジャを使った処理と同じことを SAS 言語の DATA ステッププ

ログラムで行っています。 

 

libname in POSTGSQL database=dvdrental user=XXX password=password 
schema=public; 

/* 同じ名前のテーブルを一旦削除 */ 
proc datasets lib=in; 
  delete japan_city; 
run;quit;   

/* DATAステップでDBテーブルを読み書き */ 
data in.japan_city; 
  set in.city; 
  where country_id=50; 
run;   

【重要な注意】現時点では、実行後、再度 libnameステートメントを実行してください。そ

うでないと、データセットの内容やプロパティを確認できないという問題があります。 

 

5.2. ODBC ドライバー経由の DB テーブルの入出力 

 

ODBC ドライバーは、Windows 環境や Linux 環境から各種 DB とのインタフェースを共通

方式で行うためのソフトウェアです。ほとんどの DB の ODBC ドライバーが DB ベンダや

サードパーティから提供されており、ODBCドライバーの設定を行えば、DBごとの細かい

アクセス条件等の設定を気にせずに、共通の構文で DBアクセスが実現できるという利点が

あります。 

 

［ODBC ドライバのインストール］ 

 

PostgreSQL の ODBC ドライバーは PostgreSQL の動作環境とバージョンに合わせてイン

ストールします。 もしも、次の［PC 側の ODBC ドライバーの設定］で PostgreSQL の

ODBCドライバーが見当たらない場合は、インストール環境を確認してください。 

 

［PC 側の ODBC ドライバーの設定］ 

 

WPSが導入されている PC側では、以下のような手順で、PostgreSQL の dvdrentalデータ

ベースを ODBCデータソースとして登録しておきます。 

 

コントロールパネル → システムとセキュリティ → 管理ツールで ODBC データソースを

選択します。 

（左図）「追加」をクリックします。 

（右図）リストにある PostgreSQL用 ODBCドライバーを選択し、「完了」を押します。 

 

 

（左図）データソース名と説明にはユーザーが自由に入力できます。ここでは、データソー

ス名は「PostgreSQL30」（既定のまま）、説明には「PostgreSQL dvdrental database」と入

力します。データベース名には「dvdrental」、ユーザー名とパスワードを入力します。次に

「テスト」ボタンを押し接続テストを行います。接続テストが成功することを確認すると

「保存」ボタンを押し設定終了です。 

（右図）ODBCデータソースリストにPostgreSQL用ODBCドライバーが「PostgreSQL30」

というデータソース名で追加されたことを確認します。 

［SQL 言語を使用した DB 読み取り］ 

 

SQL言語を用いてWPSから ODBCドライバー経由で PostgreSQL の DBアクセスを行う

場合は、以下のように、エンジン名として ODBC を指定し、CONNECT 文のカッコ内に 

datasrc=ODBCデータソース名などの接続オプションや dbmax_text=n（文字テキストの読

み取り最大長さの設定）などの libnameオプションを必要に応じて指定します。 

 

下記は、ODBCドライバーを用いて、dvdrentalデータベースの publicスキーマ上の cityテ

ーブルから、country_id=50 に該当するレコードのみ抽出した全カラムを含む WPS データ

セットWORK.japan_cityを作成しています。 

 

proc sql; 
  connect to ODBC (datasrc=PostgreSQL30); 
  create table japan_city as 
  select * 
  from connection to ODBC 
  ( 
    select * 
    from city 
    where country_id=50 
  ); 
  disconnect from ODBC; 
quit; 

 

(WPSワークベンチの実行ログ) 

8         proc sql; 
9           connect to ODBC (datasrc=PostgreSQL30); 
NOTE: Successfully connected to database ODBC as alias ODBC. 
10          create table japan_city as 
11          select * 
12          from connection to ODBC 
13          ( 
14            select * 
15            from public.city 
16            where country_id=50 
17          ); 
NOTE: Data set "WORK.japan_city" has 31 observation(s) and 4 variable(s) 

18          disconnect from ODBC; 
NOTE: Successfully disconnected from database ODBC. 
19        quit; 
NOTE: Procedure sql step took : 
      real time : 0.093 
      cpu time  : 0.015 

※ ODBCドライバーによる DBへのアクセス制限等の設定によっては、SQLプロシジャの

CONNECT文の（）内のオプションに、その他の指定が必要となる場合があります。以下の

ようにヘルプビューで指定可能なオプションが確認できます。 

 

ヘルプビュー13の左上にある「内容」をクリックします。 

ヘルプ全体の目次が表示されます。WPS Reference for Language Elements をクリックし

ます。 

 

 

WPSレファレンスを展開し、WPS Engine for ODBC を選択します。 

 

 

その後、展開を何度か行うと接続オプション（connection option） が表示されます。 

                                                   
13 ヘルプビューが存在しない場合は、「ウィンドウ(W)」→「ビューの表示(V)」→ 「その他(O)」→「ヘルプ」フォル

ダー → 「ヘルプ」からビューを出現させます。また、「ヘルプ(H)」→「ヘルプ目次(H)」から別ウィンドウで表示さ

れるメニューでも同じです。 

これらのオプション（user= password= datasrc= noprompt= prompt= readbuff= required=）

は ODBC ドライバー経由でのデータベースアクセスの際の CONNECT 文の DBMS オプシ

ョン指定としても、また、次に説明する libnameステートメントによるデータベースアクセ

スの接続オプションとしても用いることができ、必要に応じて指定します。 

 

［libname ステートメントによる DB 入力］ 

 

WPSのエディタから以下のプログラムを実行します。 

 

libname in ODBC datasrc=PostgreSQL30 schema=public; 

 

この指定は、WPSがデータソース PostgreSQL30に ODBCドライバー経由で接続し、ライ

ブラリ名 inで参照できるようにする指定です。（入出力両方可能） 

ただし、DBへの接続やデータテーブルの検索に必要な権限の設定や、他の接続オプション

が必要となる場合があります。 

 

さらに、libname ステートメントには schema=スキーマ名、dbmax_text=n（文字テキスト

の読み取り最大長さの設定）などの libname オプションを追加指定できます。指定可能な

libnameオプションは POSTGSQLエンジンの libnameオプションとほぼ同じですが、詳細

は前出の ODBC エンジンのヘルプ画面から確認してください。（Connection-option の下に

libname-optionが記載されています。） 

 

実行すると、ログに「ライブラリ名 in が割り当てられた」というメッセージが表示されま

す。このメッセージは ODBC経由で DBと接続できたことを表します。 

 

345       libname in ODBC datasrc=PostgreSQL30 schema=public; 
NOTE: Library in assigned as follows: 
      Engine:        ODBC 
      Physical Name: PostgreSQL30 

 

実行後、WPS サーバーエクスプローラービュー内の ローカルサーバー の下の Libraries 

を展開し、inライブラリがあることを確認してください。この中に dvdrentalデータベース

の publicスキーマ上のテーブル名およびビュー名が含まれていることがわかります。 

 

※ 右矢印がついたアイテムはテーブル名（展開すると、テーブルに含まれる項目名と項目

タイプを表すアイコンが表示されます。）、付いていないのはビュー名です。 

 

これらのテーブルやビューは、WPSデータセットに対する操作と同じファイル操作が可能

です。例えば、以下のように、テーブル名を右クリックして出現するメニューの「プロパテ

ィ」を選択すると、カラム名などを確認することができます。 

 



77 
 

 

 

 

 

※ 【重要な注意】：テーブル名をダブルクリックすると、データセットビューアが起動し、

テーブルの中身を表示しようとします。この操作はテーブルのサイズが大きいときはWPS

セッションが動かなくなったり不意に終了してしまうなどの危険を伴いますので、できる

だけ避けてください。 

 

さて、ODBC エンジンを libname ステートメントで指定した後は、以下のプログラムのよ

うに、DB テーブルやビューをWPS データセットに変換することなく、直接プロシジャ入

力に用いることもできます。この例では、customer_list ビューを入力として国（country）

別の登録顧客数の棒グラフを描いています。 

 

goptions cback=white; 
title "Customer number by Country"; 
proc gchart data=in.customer_list; 
  pie country; 
run;quit;   

※ 注意：グラフィック表示はWPS3.1.1では日本語の表示はサポートされていません。 

 

また、以下のように、SAS 言語の DATA ステップでも DB テーブルを読むことができるよ

うになります。また、SQLプロシジャを使用して処理することも可能ですので、libnameス

テートメントを使って DBアクセスを行うときは、SAS言語と SQL言語いずれも使用可能

になるというメリットがあります。 

 

下の例は、customer_list ビューから国名（country）を大文字化した値が"GREECE"のレコ

ードを選択した変数 id, name, address, city, country を含む WPS データセット

WORK.custromer_greeceを DATAステップで作成しています。 

 

data customer_greece; 
  set in.customer_list(keep=id name address city country); 
  where upcase(country)="GREECE"; 
run; 

 

［WPS データセットの DB への出力方法］ 

 

以下のいずれかの方法でWPSデータセットを DBのテーブルへ出力できます。 

 

① DBLOADプロシジャ 

② SQLプロシジャ 

③ DATAステップ 

①の方法では DB への接続方法を DBLOAD プロシジャで指定しますので、libname ステー

トメントで DBエンジンを指定する必要はありません。しかし、②と③の方法では、事前に

DB への接続に用いるエンジン名と接続先（DB 名やスキーマ名）を libname ステートメン

トで指定しておく必要があります。 

 

① DBLOADプロシジャ 

 

前節の最後のプログラムで sakiaテーブルから題名（TITLE）に"BIRD"が含まれるレコード

を選択した変数 File_id, title, rental_rate, ratingを含むWPSデータセットWORK.film_BIRD

を MySQLの film_BIRDテーブルへ出力してみましょう。以下のようにプログラムを書いて

実行します。 

 

proc dbload data=customer_greece dbms=ODBC; 
  datasrc=PostgreSQL30; 
  table=public.customer_greece; 
  load; 
run;quit; 

※ 注意： DBにアクセスするための権限や ODBCエンジンの接続オプションや libnameオ

プションがさらに必要な場合は、loadステートメントの前に、オプション名=値; の形式で

指定を追加してください。 

 

② SQLプロシジャによるテーブルへの読み書き 

 

Libnameステートメントでエンジン名を指定すると、DBテーブルをWPSデータセットと

同じように扱えるようになります。 

以下の例は sakilaデータベースの cityテーブルを読んで、country_id=50に該当するレコー

ドのみ抽出した全カラムを含むテーブル japan_cityを作成しています。 

 

libname in ODBC datasrc=PostgreSQL30 schema=public; 
proc sql; 
  create table in.japan_city as 
  select * 
  from in.city 
  where country_id = 50; 
quit; 

※ この例では、inライブラリは ODBC経由で dvdrentalデータベースの publicスキーマを

参照するよう最初の libname ステートメントで定義しています。したがって、SQL プロシ

ジャの中の inライブラリは WPSデータライブラリではなく、DBを表すものと解釈され、

読み取りも書き出しも DBテーブルが対象になります。 

③ DATAステップ 

 

以下の例は②の SQL プロシジャを使った処理と同じことを SAS 言語の DATA ステッププ

ログラムで行っています。 

 

libname in ODBC datasrc=PostgreSQL30 schema=public; 

/* 同じ名前のテーブルを一旦削除 */ 
proc datasets lib=in; 
  delete japan_city; 
run;quit;   

/* DATAステップでDBテーブルを読み書き */ 
data in.japan_city; 
  set in.city; 
  where country_id=50; 
run;   

 

 

6. SQL Server とのインタフェース 

 

［AdventureWorks2012 サンプル DB］ 

 

SQL Server 2012 Express Editionに アタッチした Microsoft提供の AdventureWorks2012

サンプル DB14を例として用います。 

SQL Server Managenent Studioで AdventureWorks2012 DBをアタッチし、内容を確認し

てください。 

 

(AdventureWorks2012データベースのテーブル一覧と Person.Addressテーブルの内容) 

                                                   
14 http://msftdbprodsamples.codeplex.com/releases/view/55330 の DOWNLOADSタブから AdventureWorks2012 

Data File をダウンロードします。 

http://msftdbprodsamples.codeplex.com/releases/view/55330%20のDOWNLOADSタブから%20AdventureWorks2012


81 
 

 

 

6.1. ネイティブドライバー経由の DB テーブルの入出力 

 

まず、コマンドラインインタフェース（CLIドライバー、ネイティブドライバー）を用いて

WPSから DBテーブルの入出力を行ってみましょう。 

ネイティブドライバーは DB と一緒にインストールされており、一般に ODBC ドライバー

より高速に DBアクセスができる点でメリットがあると言われています。 

WPS の SQLSERVR エンジンはネイティブドライバーを経由して SQL Server データベー

スとアクセスを行う仕組みを提供します。 

 

［SQL 言語を使用した DB 読み取り］ 

 

DB へのアクセスは SQL 言語を用いて行うことが多く、WPS でも SQL プロシジャを用い

て SQL言語による DBアクセスが可能です。 

 

SQL 言語を用いて WPS からネイティブドライバー経由で SQL Server の DB アクセスを

行う場合は、以下のように、エンジン名として SQLSERVR を指定し、CONNECT 文のカ

ッコ内に server=サーバー名、database=DB 名などの接続オプションや後述する

dbmax_text=n（文字テキストの読み取り最大長さの設定）などの libname オプションを指

定します。 

 

以下は AdventureWorks2012データベースの Sales.Customerテーブルから、TerritoryID=1

に該当するレコードのみ抽出した全カラムを含む WPS データセット WORK.Customer を

作成しています。 

proc sql; 
  connect to SQLSERVR(server="PC3¥SQLEXPRESS" database=AdventureWorks2012); 
  create table Customer as 
  select * 
  from connection to SQLSERVR 
  ( 
    select * 
    from Sales.Customer 
    where TerritoryID = 1; 
  ); 
  disconnect from SQLSERVR; 
quit; 

※ 3 行目の create table 文はWPS データセット WORK.Customer を作成する指定です。 

8 行目の from 節で指定した Sales.Customer は AdventureWorks2012 データベースの

salesスキーマ上の Customerテーブルを参照しています。 

 

 (WPSワークベンチの実行ログ) 

8         proc sql; 
9           connect to SQLSERVR(server="DMT-PC3¥SQLEXPRESS" database=AdventureWorks2012); 
NOTE: Successfully connected to database SQLSERVR as alias SQLSERVR. 
10          create table Customer as 
11          select * 
12          from connection to SQLSERVR 
13          ( 
14            select * 
15            from Sales.Customer 
16            where TerritoryID = 1; 
17          ); 
NOTE: Data set "WORK.Customer" has 3520 observation(s) and 7 variable(s) 
18          disconnect from SQLSERVR; 
NOTE: Successfully disconnected from database SQLSERVR. 
19        quit; 
NOTE: Procedure sql step took : 
      real time : 0.062 
      cpu time  : 0.031 

 

作成されたWPSデータセットWORK.Customerデータセットの内容を確認します。 

※ 注意：SQLSERVR ドライバーによる DB へのアクセス制限等の設定によっては、SQL

プロシジャの CONNECT 文の（）内のオプションに、その他の指定が必要となる場合があ

ります。指定可能なオプションは、以下のように、ヘルプビューで確認できます。 

 

まず、ヘルプビュー15の左上にある「内容」をクリックします。ヘルプ全体の目次が表示さ

れますので、WPS Reference for Language Elements をクリックします。 

 

WPSレファレンスを展開し、WPS Engine for SQL Server を選択します。 

 

 

Library Engines → SQLSERVR を選択すると、データベース接続オプション（connection-

option）が表示されます。 

                                                   
15 ヘルプビューが存在しない場合は、「ウィンドウ(W)」→「ビューの表示(V)」→ 「その他(O)」→「ヘルプ」フォル

ダー → 「ヘルプ」からビューを出現させます。また、「ヘルプ(H)」→「ヘルプ目次(H)」から別ウィンドウで表示さ

れるメニューでも同じです。 

これらの接続オプション（user= password= server= database= readbuff= port= driver=）は

SQLSERVRドライバー経由でのデータベースアクセスの際の CONNECT文の DBMSオプ

ション指定としても、また、次に節で説明する libnameステートメントによる DB接続オプ

ションとしても用いることができ、必要に応じて指定します。 

 

［libname ステートメントによる DB 入力］ 

 

WPSのエディタから以下のプログラムを入力し、実行します。 

 

libname person SQLSERVR server="PC3¥SQLEXPRESS" database=AdventureWorks2012 
schema=Person; 
libname sales SQLSERVR server="PC3¥SQLEXPRESS" database=AdventureWorks2012 
schema=Sales; 

 

この指定は、WPSがデータベース AdventureWorks2012に SQLSERVRネイティブドライ

バー経由で接続し、Personスキーマ上のテーブルとビューをライブラリ名 personで、Sales

スキーマ上のテーブルとビューをライブラリ名 sales で、それぞれ参照できるようにする

指定です。（入出力両方可能） ただし、DBへの接続やデータテーブルの検索に必要な権限

の設定や、他の接続オプションが必要となる場合があります。 

 

なお、server=サーバー名、database=DB名指定は接続オプション、schema=指定は libname

オプションです。libnameオプションには他に、dbmax_text=n（文字テキストの読み取り最

大長さの設定）などがあり、以下のような libnameオプションが指定可能です。 

さて、実行すると、ログに person と sales の 2 個の「ライブラリ名が割り当てられた」

というメッセージが表示されます。このメッセージは SQLSERVR エンジン経由で DBと接

続できたことを表します。 

 

29        libname person SQLSERVR server="DMT-PC3¥SQLEXPRESS" database=AdventureWorks2012 schema 
29      ! =Person; 
NOTE: Library person assigned as follows: 
      Engine:        SQLSERVR 
      Physical Name: AdventureWorks2012 
 
30        libname sales SQLSERVR server="DMT-PC3¥SQLEXPRESS" database=AdventureWorks2012 schema= 
30      ! Sales; 
NOTE: Library sales assigned as follows: 
      Engine:        SQLSERVR 
      Physical Name: AdventureWorks2012 

 

実行後、WPS サーバーエクスプローラービュー内の ローカルサーバー の下の Libraries 

を展開し、Person ライブラリと Sales ライブラリがあることを確認してください。これら

の中に AdventureWorks2012データベースの Personスキーマ、Salesスキーマ上のテーブ

ル名およびビュー名がそれぞれ含まれていることがわかります。 

 

※ 右矢印がついたアイテムはテーブル名（展開すると、テーブルに含まれる項目名と項目

タイプを表すアイコンが表示されます。）、付いていないのはビュー名です。 

 

これらのテーブルやビューは、WPSデータセットに対する操作と同じファイル操作が可能

です。例えば、以下のように、テーブル名を右クリックして出現するメニューの「プロパテ

ィ」を選択すると、カラム名などを確認することができます。 

※ 【重要な注意】：テーブル名をダブルクリックすると、データセットビューアが起動し、

テーブルの中身を表示しようとします。この操作はテーブルのサイズが大きいときはWPS

セッションが動かなくなったり不意に終了してしまうなどの危険を伴いますので、できる

だけ避けてください。 

 

さて、SQLSERVR エンジンを libnameステートメントで指定した後は、以下のプログラム

のように、DB テーブルやビューをWPS データセットに変換することなく、直接プロシジ

ャ入力に用いることもできます。この例では、Sales.Salespersonテーブルを入力としてビ

ジネスエリア別の売り上げ高の円グラフを描いています。 

 

goptions cback=white; 
title "Pie Chart of SalesYTD by SalesPersonID";  
proc gchart data=sales.salesperson; 
  pie businessentityid/sumvar=salesYTD; 
run; 

※ 注意：グラフィック表示はWPS3.1.1では日本語の表示はサポートされていません。 

 

また、以下のように、SAS 言語の DATA ステップでも DB テーブルを読むことができるよ

うになります。また、SQLプロシジャを使用して処理することも可能ですので、libnameス

テートメントを使って DBアクセスを行うときは、SAS言語と SQL言語いずれも使用可能

になるというメリットがあります。 

 

下の例は、Person.Address テーブルから 3 つの都市名のいずれかに該当するレコードを選

択したWPSデータセットWORK.address_subsetを DATAステップで作成しています。 

data address_subset; 
  set person.address(keep=AddressID AddressLine1 City); 
  where City in ('San Francisco', 'Los Angeles', 'Seattle'); 
run; 

 

［WPS データセットの DB への出力方法］ 

 

以下のいずれかの方法でWPSデータセットを DB のテーブルへ出力できます。 

 

① DBLOADプロシジャ 

② SQLプロシジャ 

③ DATAステップ 

 

①の方法では DB への接続方法を DBLOAD プロシジャで指定しますので、libname ステー

トメントで DBエンジンを指定する必要はありません。しかし、②と③の方法では、事前に

DB への接続に用いるエンジン名と接続先（DB 名やスキーマ名）を libname ステートメン

トで指定しておく必要があります。 

 

① DBLOADプロシジャ 

 

前節の最後のプログラムで Person.Addressテーブルから 3つの都市名のいずれかに該当す

るレコードを選択したWPSデータセットWORK.address_subsetをSQL Server のPerson

スキーマ上の Address_Subsetテーブルへ出力してみましょう。以下のようにプログラムを

書いて実行します。 

 

proc dbload data=address_subset dbms=SQLSERVR; 
  server="PC3¥SQLEXPRESS"; 
  database=AdventureWorks2012; 
  table=Person.Address_Subset; 
  load; 
run;quit; 

※ 注意： DBにアクセスするため SQLSERVRエンジンの接続オプションや libnameオプ

ションがさらに必要な場合は、loadステートメントの前に、オプション名=値; の形式で指

定を追加してください。 

 

② SQLプロシジャによるテーブルへの読み書き 

 

Libnameステートメントでエンジン名を指定すると、DBテーブルをWPSデータセットと

同じように扱えるようになります。 

以下の例は AdventureWorks2012 データベースの Sales.Customer テーブルを読んで、

TerritoryID=1 に 該 当 す る レ コ ー ド の み 抽 出 し た 全 カ ラ ム を 含 む テ ー ブ ル

Sales.Customer_Territory_1を作成しています。 

 

libname sales SQLSERVR server="PC3¥SQLEXPRESS" database=AdventureWorks2012 
schema=Sales; 
proc sql; 
  create table Sales.Customer_Territory_1 as 
  select * 
  from Sales.Customer 
  where TerritoryID = 1; 
quit; 

※ この例では、Salesライブラリは SQL Server DBの Salesスキーマを参照するよう最初

の libname ステートメントで定義しています。したがって、SQL プロシジャの中の Sales

ライブラリは WPSデータライブラリではなく、AdventureWorks2012 DBの Salesスキー

マを表すものと解釈され、読み取りも書き出しも DBテーブルが対象になります。 

③ DATAステップ 

 

以下の例は②の SQL プロシジャを使った処理と同じことを SAS 言語の DATA ステッププ

ログラムで行っています。 

 

libname sales SQLSERVR server="PC3¥SQLEXPRESS" database=AdventureWorks2012 
schema=Sales; 

/* 同じ名前のテーブルを一旦削除 */ 
proc datasets lib=sales; 
  delete Customer_Territory_1; 
run;quit;   

/* DATAステップでDBテーブルを読み書き */ 
data Sales.Customer_Territory_1; 
  set Sales.Customer; 
  where TerritoryID = 1; 
run;   

 

6.2. ODBC ドライバー経由の DB テーブルの入出力 

 

ODBC ドライバーは、Windows 環境や Linux 環境から各種 DB とのインタフェースを共通

方式で行うためのソフトウェアです。ほとんどの DB の ODBC ドライバーが DB ベンダや

サードパーティから提供されており、ODBCドライバーの設定を行えば、DBごとの細かい

アクセス条件等の設定を気にせずに、共通の構文で DBアクセスが実現できるという利点が

あります。 

 

［ODBC ドライバのインストール］ 

 

DB ベンダーのサイトから ODBC ドライバーをダウンロードし、インストールしておきま

す。ここでは、Microsoftのサイトから SQL Serverの ODBCドライバーをダウンロードし

ています。 

［PC 側の ODBC ドライバーの設定］ 

 

WPS が導入されている PC 側では、以下のような手順で、上記 SQL Server 2012 の

AdventureWorks2012データベースを ODBCデータソースとして登録しておきます。 

 

コントロールパネル → システムとセキュリティ → 管理ツールで ODBC データソース

を選択します。 

 

 

（左図）「追加」をクリックします。 

（右図）ダウンロードした SQL Server用 ODBCドライバーを選択し、「完了」を押します。 

 

 

（左図）新規データソースの名前と説明にはユーザーが自由に入力できます。ここではわか

りやすく、名前に「AdventureWorks」、説明には「SQL Server Sample Database」と入力

を行いました。「接続する SQL Serverを選択してください」には、SQL Serverのサーバー

名を入力します。「次へ(N)」をクリックします。 

（右図）ログイン方法の設定を確認して「次へ(N)」をクリックします。 

（左図）この画面で「既定のデータベースを以下に変更する」にチェックを入れて、アクセ

スしたいデータベース名（ここでは AdventureWorks2012）をプルダウンリストから選択し、

「次へ(N)」をクリックします。 

（右図）そのまま「完了」をクリックします。 

 

 

（左図）「データソースのテスト」をクリックし接続テストを行います。 

（右図）テスト結果を確認し、「OK」をクリックし、戻ります。 

 

 

AdventureWorksが ODBC データソースに追加されたことを確認し、「OK」を押して設定終

了です。 

［SQL 言語を使用した DB 読み取り］ 

 

SQL 言語を用いてWPS から ODBC ドライバー経由で SQL Server の DBアクセスを行う

場合は、以下のように、エンジン名として ODBC を指定し、CONNECT 文のカッコ内に

datasrc=ODBCデータソース名などの接続オプションや dbmax_text=n（文字テキストの読

み取り最大長さの設定）などの libnameオプションを指定します。 

 

以下は、ODBCドライバーを用いて、AdventureWorks2012データベースの Person.Address

テーブルの 3つのカラム（AddressID, AddressLine1, City）を Cityの値が 'San Francisco' 

か 'Los Angeles' か  'Seattle' のいずれかに該当するレコードを抽出し、AddressIDの小さ

い方から並べたWPSデータセットWORK.Aを作成する例となっています。 

 

proc sql; 
  connect to ODBC(datasrc=AdventureWorks); 
  create table A as 
  select *  
  from connection to ODBC 
  ( 
    select AddressID, AddressLine1, City  
    from Person.Address  
    where City in ('San Francisco', 'Los Angeles', 'Seattle') 
    order by AddressID 
  ); 
  disconnect from ODBC; 
quit;   

 

(WPSワークベンチの実行ログ) 

95        proc sql; 
96          connect to ODBC(datasrc=AdventureWorks); 
NOTE: Successfully connected to database ODBC as alias ODBC. 



94 
 

97          create table A as 
98          select * 
99          from connection to ODBC 
100         ( 
101           select AddressID, AddressLine1, City 
102           from Person.Address 
103           where City in ('San Francisco', 'Los Angeles', 'Seattle') 
104           order by AddressID 
105         ); 
NOTE: Data set "WORK.A" has 285 observation(s) and 3 variable(s) 
106         disconnect from ODBC; 
NOTE: Successfully disconnected from database ODBC. 
107       quit; 
NOTE: Procedure sql step took : 
      real time : 0.046 
      cpu time  : 0.015 

※ ODBCドライバーによる DBへのアクセス制限等の設定によっては、SQLプロシジャの

CONNECT文の（）内のオプションに、その他の指定が必要となる場合があります。以下の

ようにヘルプビューで指定可能なオプションが確認できます。 

 

ヘルプビュー16の左上にある「内容」をクリックします。 

ヘルプ全体の目次が表示されます。WPS Reference for Language Elements をクリックし

ます。 

 

 

WPSレファレンスを展開し、WPS Engine for ODBC を選択します。 

                                                   
16 ヘルプビューが存在しない場合は、「ウィンドウ(W)」→「ビューの表示(V)」→ 「その他(O)」→「ヘルプ」フォル

ダー → 「ヘルプ」からビューを出現させます。また、「ヘルプ(H)」→「ヘルプ目次(H)」から別ウィンドウで表示さ

れるメニューでも同じです。 

その後、展開を何度か行うと接続オプション（connection option） が表示されます。 

 

 

これらのオプション（user= password= datasrc= noprompt= prompt= readbuff= required=）

は ODBC ドライバー経由でのデータベースアクセスの際の CONNECT 文の DBMS オプシ

ョン指定としても、また、次に説明する libnameステートメントによるデータベースアクセ

スの接続オプションとしても用いることができ、必要に応じて指定します。 

 

［libname ステートメントによる DB 入力］ 

 

WPSのエディタから以下のプログラムを実行します。 

 

libname person ODBC datasrc=AdventureWorks schema=Person; 
libname sales ODBC datasrc=AdventureWorks schema=Sales; 

 

この指定は、WPS がデータソース AdventureWorks に ODBC ドライバー経由で接続し、

Person スキーマをライブラリ名 person で、Sales スキーマのテーブルをライブラリ名 

salesで、それぞれ参照できるようにする指定です。（入出力両方可能） 

なお、datasrc=指定は DB接続オプション、shema=指定は libnameオプションです。 

libnameオプションには他に、dbmax_text=n（文字テキストの読み取り最大長さの設定）な

どがあります。指定可能な libname オプションは SQLSERVR エンジンの libname オプシ

ョンとほぼ同じですが、詳細は前出の ODBCエンジンのヘルプ画面から確認してください。

（Connection-optionの下に libname-optionが記載されています。） 

 

実行すると、ログに person と sales 2個の「ライブラリ名が割り当てられた」というメッ

セージが表示されます。このメッセージはODBC経由でDBと接続できたことを表します。 

 

128       libname person ODBC datasrc=AdventureWorks schema=Person; 
NOTE: Library person assigned as follows: 
      Engine:        ODBC 
      Physical Name: AdventureWorks 
 
129       libname sales ODBC datasrc=AdventureWorks schema=Sales; 
NOTE: Library sales assigned as follows: 
      Engine:        ODBC 
      Physical Name: AdventureWorks 

 

実行後、WPS サーバーエクスプローラービュー内の ローカルサーバー の下の Libraries 

を展開し、Person ライブラリと Sales ライブラリがあることを確認してください。これら

の中に AdventureWorks2012データベースの Personスキーマ、Salesスキーマ上のテーブ

ル名およびビュー名がそれぞれ含まれていることがわかります。 

※ 右矢印がついたアイテムはテーブル名（展開すると、テーブルに含まれる項目名と項目

タイプを表すアイコンが表示されます。）、付いていないのはビュー名です。 

 

これらのテーブルやビューは、WPSデータセットに対する操作と同じファイル操作が可能

です。例えば、以下のように、テーブル名を右クリックして出現するメニューの「プロパテ

ィ」を選択すると、カラム名などを確認することができます。 

※ 【重要な注意】：テーブル名をダブルクリックすると、データセットビューアが起動し、

テーブルの中身を表示しようとします。この操作はテーブルのサイズが大きいときはWPS

セッションが動かなくなったり不意に終了してしまうなどの危険を伴いますので、できる

だけ避けてください。 

 

さて、ODBC アクセスエンジンを libname ステートメントで指定すると、以下のプログラ

ムのように、DB テーブルやビューをWPS データセットに変換することなく、直接プロシ

ジャ入力に用いることもできます。この例は、sales.salesperson テーブルを入力としてビ

ジネスエリア別の売り上げ高の円グラフを描いています。 

 

goptions cback=white; 
title "Pie Chart of SalesYTD by SalesPersonID";  
proc gchart data=sales.salesperson; 
  pie businessentityid/sumvar=salesYTD; 
run; 

 

 

※ 注意：グラフィック表示はWPS3.1.1では日本語の表示はサポートされていません。 



99 
 

 

また、以下のように、SAS 言語の DATA ステップでも DB テーブルを処理することができ

るようになります。また、SQLプロシジャを使用して処理することも可能ですので、libname

ステートメントを使って DBアクセスを行うときは、SAS言語と SQL言語いずれも使用可

能になるというメリットがあります。 

 

下の例は、前の節で SQL プロシジャで作成した WORK.A と同じ内容のデータセット

WORK.Bを DATAステップで作成しています。 

data B; 
  set person.address(keep=AddressID AddressLine1 City); 
  where City in ('San Francisco', 'Los Angeles', 'Seattle'); 
run; 

 

［WPS データセットの DB への出力方法］ 

 

以下のいずれかの方法でWPSデータセットを DB のテーブルへ出力できます。 

 

① DBLOADプロシジャ 

② SQLプロシジャ 

③ DATAステップ 

 

①の方法では DB への接続方法を DBLOAD プロシジャで指定しますので、libname ステー

トメントで DBエンジンを指定する必要はありません。しかし、②と③の方法では、事前に

DBへの接続方法を libnameステートメントで指定しておく必要があります。 

 

① DBLOADプロシジャ 

 

前節の最後のプログラムで Person.Addressテーブルから 3つの都市名のいずれかに該当す

るレコードを選択した WPS データセット WORK.A を SQL Server の Person スキーマ上

の Address_Subset_A テーブルへ出力してみましょう。以下のようにプログラムを書いて

実行します。 

 

proc dbload data=A dbms=ODBC; 
  datasrc=AdventureWorks; 
  table=Person.Address_Subset_A; 
  load; 
run; 

※ 注意： DBにアクセスするための権限や ODBCエンジンの接続オプションや libnameオ

プションがさらに必要な場合は、loadステートメントの前に、オプション名=値; の形式で

指定を追加してください。 

 

② SQLプロシジャによるテーブルへの読み書き 

 

Libnameステートメントで ODBCエンジンを指定すると、DBテーブルをWPSデータセッ

トと同じように扱えるようになります。 

以下の例は AdventureWorks2012 データベースの Sales.Customer テーブルを読んで、

TerritoryID=1 に 該 当 す る レ コ ー ド の み 抽 出 し た 全 カ ラ ム を 含 む テ ー ブ ル

Sales.Customer_Territory_1を作成しています。 

 

libname sales ODBC datasrc=AdventureWorks schema=Sales; 
proc sql; 
  create table Sales.Customer_Territory_1 as 
  select * 
  from Sales.Customer 
  where TerritoryID = 1; 
quit; 

※ この例では、Salesライブラリは SQL Server DBの Salesスキーマを参照するよう最初

の libname ステートメントで定義しています。したがって、SQL プロシジャの中の Sales

ライブラリは WPSデータライブラリではなく、AdventureWorks2012 DBの Salesスキー

マを表すものと解釈され、読み取りも書き出しも DBテーブルが対象になります。 

 

③ DATAステップ 

 

以下の例は②の SQL プロシジャを使った処理と同じことを SAS 言語の DATA ステッププ

ログラムで行っています。 

 

libname sales ODBC datasrc=AdventureWorks schema=Sales; 

/* 同じ名前のテーブルを一旦削除 */ 
proc datasets lib=sales; 
  delete Customer_Territory_1; 
run;quit;   

/* DATAステップでDBテーブルを読み書き */ 
data Sales.Customer_Territory_1; 
  set Sales.Customer; 
  where TerritoryID = 1; 
run;   

 

 

以上 

 


